



SETLX — A Tutorial

Version 1.3

Karl Stroetmann

Duale Hochschule Baden-Württemberg Stuttgart

`stroetmann@dhbw-stuttgart.de`

Tom Herrmann

Duale Hochschule Baden-Württemberg Stuttgart

`work@herrmann-tom.de`

August 28, 2012

Abstract

In the late sixties, Jack Schwartz, a renowned professor of mathematics at the Courant Institute for Mathematical Sciences in New York, developed a very high level programming language called SETL [Sch70, SDSD86]. The most distinguishing feature of this language is the support it offers for sets and lists. This feature permits convenient implementations of mathematical algorithms: As set theory is the language of mathematics, many mathematical algorithms that are formulated in terms of set theory have very straightforward implementations in SETL.

Unfortunately, at the time of its invention, SETL did not get the attention that it deserved. One of the main reasons was that SETL is an interpreted language and in those days, the run time overhead of an interpreter loop was not yet affordable. More than forty years after the first conception of SETL, the efficiency of computers has changed dramatically and for many applications, the run time efficiency of a language is no longer as important as it once was. After all, modern scripting languages like Python [vR95] or Ruby [FM08] are all interpreted and noticeably slower than compiled languages like C, but this fact hasn't impeded their success.

At the Baden-Württemberg Corporate State University, the first author has used SETL2 [Sny90] for some years in a number of his introductory computer science courses. He has noticed that the adoption of SETL has made the abstract concepts of set theory tangible to the students. Nevertheless, as the original version of SETL is more than forty years old, it has a number of shortcomings. One minor issue is the fact, that the syntax is quite dated and has proven difficult to master for students that are mainly acquainted with C and Java. Furthermore, SETL lacks any immediate support for first order terms. Therefore, SETL has been extended into the new language SETLX. The main features that have been changed or added are as follows:

- SETLX supports terms in a way similar to the language *Prolog*. In particular, SETLX supports matching.
- SETLX supports some ideas from functional programming. In particular, functions can be used as a primitive data type. Furthermore, SETLX supports memoization for functions.
- SETLX provides some means to support backtracking.
- Lastly, while SETL has a syntax that is reminiscent of Algol, SETLX has a syntax that is more akin to languages like C or Java.

The language SETLX has been implemented by Tom Herrmann as part of his student research project.

Contents

1	Introduction	4
2	SetlX Types	6
2.1	Getting Started	6
2.2	Boolean Values	9
2.3	Sets	10
2.3.1	Operators on Sets	11
2.3.2	Set Comprehensions	13
2.3.3	Miscellaneous Set Functions	14
2.4	Lists	15
2.5	Pairs, Relations, and Functions	16
2.6	Strings	18
2.7	Terms	19
3	Writing Statements in SetlX	22
3.1	Assignment Statements	22
3.2	Functions	23
3.2.1	Memoization	24
3.3	Branching Statements	26
3.3.1	<code>if-then-else</code> Statements	26
3.3.2	<code>switch</code> Statements	27
3.4	Matching	28
3.4.1	String Matching	28
3.4.2	List Matching	30
3.4.3	Set Matching	30
3.4.4	Term Matching	30
3.5	Loops	33
3.5.1	<code>while</code> Loops	33
3.5.2	<code>for</code> Loops	34
4	Exceptions and Backtracking	37
4.1	Exceptions	37
4.1.1	Different Kinds of Exceptions	39
4.2	Backtracking	40
5	Predefined Functions	45
5.1	Functions and Operators on Sets and Lists	45
5.2	Functions for String Manipulation	47
5.3	Functions for Term Manipulation	49
5.4	Mathematical Functions	51
5.5	Generating Random Numbers and Permutations	53
5.5.1	<code>shuffle</code>	54

5.5.2	nextPermutation	55
5.5.3	permutations	55
5.6	Type Checking Functions	55
5.7	Interactive Debugging	56
5.8	I/O Functions	57
5.8.1	appendFile	58
5.8.2	deleteFile	58
5.8.3	get	58
5.8.4	load	58
5.8.5	loadLibrary	58
5.8.6	multiLineMode	59
5.8.7	nPrint	59
5.8.8	nPrintErr	59
5.8.9	print	59
5.8.10	printErr	59
5.8.11	read	59
5.8.12	readFile	60
5.8.13	writeFile	60
5.9	Miscellaneous Functions	60
5.9.1	abort	60
5.9.2	cacheStats	60
5.9.3	clearCache	61
5.9.4	compare	61
5.9.5	getScope	61
5.9.6	logo	62
5.9.7	now	62
5.9.8	sleep	62

Chapter 1

Introduction

Every year, dozens of new programming languages are proposed and each time this happens, the inventors of the new language have to answer the same question: There are hundreds of programming languages already, why do we need yet another one? Of course, the answer is always the same. It consists of an economical argument, a theological argument and a practical argument. For the convenience of the reader, let us briefly review these arguments as they read for SETLX.

1. Nothing less than the prosperity and welfare of the entire universe is at stake and only SETLX provides the means to save it.
2. Programming in SETLX is the only way to guarantee redemption from the eternal hell fire that awaits those not programming in SETLX.
3. Programming in SETLX is fun!

The economical argument has already been discussed at length by Adams [Ada80], therefore we don't have to repeat it here. We deeply regret the fact that the background of the average computer scientist does not permit them to follow advanced theological discussions. Therefore, we refrain from giving a detailed proof of the second claim. Nevertheless, we hope the examples given in this tutorial will convince the reader of the truth of the third claim. One of the reasons for this is that SETLX programs are both very concise and readable. This often makes it possible to fit the implementation of complex algorithms in SETLX on a single slide because the SETLX program is as concise as the pseudocode that is usually used to present algorithms in lectures. The benefit of this is that instead of pseudocode, students have a running program. Indeed, the conciseness of SETLX programs was one of the reasons for the first author to adopt SETLX as a programming language in his courses on computer science: It is often feasible to write a complete SETLX program in a few lines onto the blackboard, since SETLX programs are as concise as mathematical formulae.

SETLX is well suited to implement complex algorithms. This is achieved because SetlX provides a number of sophisticated builtin data types that enable the user to code at a very high abstraction level. These types are sets, lists, first-order terms, and functions. The purpose of this tutorial is to introduce the most important features of SETLX and to show how the use of the above mentioned data types leads to programs that are both shorter and clearer than the corresponding programs in other programming languages. The remainder of this tutorial is structured as follows:

1. In the second chapter, we discuss the available data types.
2. The third chapter shows the control structures available.
3. The fourth chapter discusses the **try-catch** and **throw** mechanism and demonstrates how this mechanism can be used to simulate backtracking.
4. The final chapter explains the predefined functions.

This tutorial is not meant as an introduction to programming. It assumes that the reader has had some preliminary exposure to programming and has already written a few programs in either C, Java, or a similar language.

Downloading

The current distribution of SETLX can be downloaded from either

<http://www.lehre.dhbw-stuttgart.de/~stroetma/SetlX/setlX.php>

or

<http://randoom.org/Software/SetlX>.

SETLX is written in Java and is therefore supported on a number of different operating systems. Currently, SETLX is supported on *Linux*, *Mac OS X*, *Microsoft Windows*, and *Android*.

The websites given above explain how to install the language on various platforms. The distribution contains the Java code and a development guide that gives an overview of the implementation.

Disclaimer

The development of SETLX is an ongoing project. Therefore some of the material presented in this tutorial might be out of date, while certain aspects of the language won't be covered. The current version of this tutorial is not intended to be a reference manual. The idea is rather to provide the reader with an introduction that is sufficient to get started.

Encouragement

The authors would be grateful for any kind of feedback. The authors can be contacted via email as follows:

Karl Stroetmann: stroetmann@dhbw-stuttgart.de

Tom Herrmann: work@herrmann-tom.de

Acknowledgements

The authors would like to acknowledge that Karl-Friedrich Gebhardt and Hakan Kjellerstrand have both read earlier drafts of this tutorial and have given valuable feedback that has helped to improve the current presentation.

Chapter 2

SetlX Types

This chapter contains a short overview of the data types supported by SETLX and tries to whet your appetite for the language by showing off some of the features that are unique to SETLX. However, before we discuss the more elaborate data types, we introduce the basic data types for numbers and strings and show how to invoke the interpreter.

2.1 Getting Started

SETLX is an interpreted language. To start the interpreter, the file `setlX` has to be both executable and part of the search path. If these preconditions are satisfied, the command

```
setlX
```

launches the interpreter. The interpreter first prints the banner shown in Figure 2.1, followed by a prompt `=>`. Commands are typed after the prompt. If the command is an assignment or an expression, then it is terminated by a semicolon. However, complex commands like, for example, loops are not terminated by a semicolon.

```
=====setlX=====v1.3.0==

Welcome to the setlX interpreter!

Open Source Software from http://setlX.randoom.org/
(c) 2011-2012 by Herrmann, Tom

You can display some helpful information by using '--help' as parameter when
launching this program.

Interactive-Mode:
  The 'exit;' statement terminates the interpreter.

=====Interactive=Mode=====

=>
```

Figure 2.1: The SETLX banner followed by a prompt.

The SETLX interpreter can be used as a simple calculator: Typing

```
=> 1/3 + 2/5;
```

and then hitting the return key **once**¹ will result in the following response:

```
~< Result: 11/15 >~
```

After printing the result, the interpreter provides a new prompt so the next command can be entered. Incidentally, the last example shows the first data type supported by SETLX: rational numbers. A rational number consists of a nominator and a denominator, both of which are integers. SETLX takes care to ensure that nominator and denominator are always in lowest terms. Furthermore, if the denominator has the value 1, only the nominator is printed. Therefore, after typing

```
=> 1/3 + 2/3;
```

SETLX will respond:

```
~< Result: 1 >~
```

The precision of rational numbers is only limited by the available memory. For example, the command

```
=> 50!;
```

computes the factorial of 50 and yields the result

```
~< Result: 30414093201713378043612608166064768844377641568960512000000000000 >~.
```

If you prefer to calculate with floating point values, the easiest way is to add 0.0 at the end of the expression because if an arithmetic expression contains a floating point value, the result is converted to a floating point number. Therefore, the command

```
=> 1/3 + 2/5 + 0.0;
```

yields the answer:

```
~< Result: 0.7333333333333333 >~
```

Of course, the same can also be achieved via the command

```
=> 1.0/3 + 2/5;
```

Since the precision of rational numbers in SETLX is not limited, we can do things like computing $\sqrt{2}$ to a hundred decimal places. Figure 2.2 on page 8 shows a program that computes $\sqrt{2}$ as the limit of the sequence $(b_n)_n$ that is defined as

$$b_0 := 2 \quad \text{and} \quad b_{n+1} := \frac{1}{2} \cdot \left(b_n + \frac{2}{b_n} \right).$$

In order to print a rational number in decimal notation with a fixed number of places, the function `nDecimalPlaces(x, n)` has been used. The first argument x is the rational number to be printed, while the second argument n gives the number of places to be printed. To run this program from the command line, assume it is stored in a file with the name `sqrt.stlx` in the current directory. Then the command

```
setlX sqrt.stlx
```

loads and executes this program. Alternatively, the command can be executed interactively in the interpreter via the following command:

```
=> load("sqrt.stlx");
```

¹ In SETLX, statements can extend over many lines. If the user intends to use multiline statements, then she can start the interpreter using the commandline switch “`--multiLineMode`”. In multi line mode, the return key needs to be hit twice to signal the end of the input. Instead of using a commandline switch, the user can also issue the command

```
multiLineMode(true);
```

to activate multiline mode. To switch back to single line mode, use the command “`multiLineMode(false);`”.

The output of the program is shown in Figure 2.3.

```

1  b := 2;
2  for (n in [1 .. 9]) {
3      b := 1/2 * (b + 2/b);
4      print(n + ": " + nDecimalPlaces(b, 100));
5  }

```

Figure 2.2: A program to calculate $\sqrt{2}$ to 100 places.

[illegible]

Figure 2.3: The output produced by the program in Figure 2.2.

The last program also demonstrates that SETLX supports strings. In SETLX, any sequence of characters enclosed in double quote characters is a string. For example, the *hello world program* in SETLX is just

```
=> "Hello world!";
```

It yields the output:

```
~< Result:  "Hello world!" >~
```

However, this only works in interactive mode. If you want to be more verbose² or if you are not working interactively, you can instead write

```
=> print("Hello world!");
```

This will yield two lines of output:

```
Hello world!  
~< Result: om >~
```

The first lines show the effect of the invocation of the function `print`, the second line gives the return value computed by the call of the function `print()`. As the function `print()` does not return a meaningful value, the return value is the *undefined value* Ω . In SETLX, this value is written as `om`.

In order to assign the value of an expression to a variable, SETLX provides the assignment operator “:=”. Syntactically, this operator is the only major deviation from the syntax of the programming language C. For example, the statement

$$\Rightarrow x := 1/3;$$

binds the variable `x` to the value `1/3`.

Important: In SETLX, the names of variables have to start with a lower case letter. After the first letter, the name can contain upper or lower case letters, digits, or underscores. If the first letter is an upper case letter, then the identifier is interpreted as a *functor*. This notion will be explained when discussing *terms*.

² If you want to be much more verbose, you should program in either *Cobol* or *Java*.

2.2 Boolean Values

The Boolean values **true** and **false** represent truth and falsity. Boolean expressions can be constructed using the comparison operators “==”, “!=”, “<”, “>”, “<=”, and “>=”. SETLX provides the following propositional operators to combine Boolean expressions.

1. “&&” denotes the logical *and* (also known as *conjunction*), so the expression

a && b

is true if and only if both **a** and **b** are true.

2. “||” denotes the logical *or*, (also known as *disjunction*), so the expression

a || b

is true if either **a** or **b** is true. It is also true if both **a** and **b** is true.

3. “!” denotes the logical *not*, (also known as *negation*), so the expression

!a

is true if and only if **a** is false.

4. “=>” denotes the logical *implication*, so the expression

a => b

is true if either **a** is false or **b** is true. Therefore, the expression

a => b

has the same value as the expression

!a || b.

5. “<==>” denotes the logical *equivalence*, so the expression

a <==> b

is true if either **a** and **b** are both true or **a** and **b** are both false. Therefore, the expression

a <==> b

has the same value as the expression

(a && b) || (!a && !b).

6. “<!=>” denotes the logical *antivalence*, so the expression

a <!=> b

is true if the truth values of **a** and **b** are different. Therefore, the expression

a <!=> b

is equivalent to the expression

!(a <==> b).

The operators “==” and “!=” can be used instead of the operators “<==>” and “<!=>”.

Furthermore, SETLX supports both the universal quantifier “forall” and the existential quantifier “exists”. For example, to test whether the formula

$$\forall x \in \{1, \dots, 10\} : x^2 \leq 2^x$$

is true we can evaluate the following expression:

forall (x in {1..10} | x ** 2 <= 2 ** x);

This expression checks whether x^2 less than or equal to 2^x for all x between 1 and 10. Syntactically, a `forall` expression is described by the following grammar rule:

$$expr \rightarrow \text{"forall" "(" var "in" setExpr "|" cond ")"}$$

Here, `var` denotes a variable and `expr` is an expression that evaluates to a set s (or a list or a string), while `cond` is a Boolean expression. The `forall` expression evaluates to `true` if for all elements of s the condition `cond` evaluates to true. There is a generalization of the grammar rule that allows to check several variables simultaneously, so we can write an expression like the following:

```
forall (x in {1..10}, y in [20..30] | x < y).
```

If instead we want to know whether the formula

$$\exists x \in \{1, \dots, 10\} : 2^x < x^2$$

is true, we have to write:

```
exists (x in {1..10} | 2 ** x < x ** 2);
```

This expression checks whether there exists a natural number $x \in \{1, \dots, 10\}$ such that 2^x is less than x^2 .

2.3 Sets

The most interesting data type provided by SETLX is *set* type. A *set* is a collection that contains all of its elements exactly once. To create a simple set containing the numbers 1, 2, and 3, we can write

```
=> {1, 2, 3};
```

Note that the comparison

```
=> {1,2,3} == {2,3,1}
```

yields the result

```
Result: true
```

After all, the order of elements in a set does not matter. This behaviour can lead to results that are sometimes surprising for the novice. For example, the command

```
=> print({3,2,1});
```

yields the following output:

```
{1, 2, 3}
```

The reason is, that SETLX does not remember the ordering of elements in a set, it just remembers the elements and sorts them internally to maximize the efficiency of looking up elements from a set.

To check whether a set contains a given entity as an element, SETLX provides the binary infix operator `in`. For example, the command

```
=> 2 in {1,2,3};
```

yields the result `true`, as 2 is indeed an element of the set $\{1, 2, 3\}$, while

```
=> 4 in {1,2,3};
```

returns `false`.

The data type of a set would be quite inconvenient to use if we could only create sets by explicitly listing all their elements. Fortunately, there are some more powerful expressions to create sets. The most straightforward of them is the *range operator* that can create a set containing all

the integers in a given range. For example, to create the set containing the integers from 1 up to 16, we can write

=> { 1..16 };

This will give the result

Result: {1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16}.

Here, we have used the range operator “..”. For given integers a and b such that $a \leq b$, the expression

{ $a..b$ }

will generate the set of all integers starting from a up to and including the number b . Mathematically, the semantics of this operator is given by the formula

$$\{ a..b \} = \{ a + n \mid a \in \mathbb{N} \wedge a + n \leq b \}.$$

If b is less than a , the expression

{ $a..b$ }

denotes the empty set {}.

While neighbouring elements of a set created by the expression “{ $a..b$ }” differ by 1, there is a variant of the range operator that allows us to specify the size of the difference between neighbouring elements. For example, the expression

=> {1,3..10};

yields

Result: {1, 3, 5, 7, 9}.

In general, when a set definition of the form

$a, b..c$

is evaluated, there are two cases.

1. $a < b$. In this case, the *step size* s is defined as

$$s := b - a.$$

Then, we have

$$\{ a, b..c \} := \{ a + n \cdot s \mid n \in \mathbb{N} \wedge a + n \cdot s \leq c \}.$$

2. $a > b$. In this case, the *step size* s is defined as

$$s := a - b.$$

Then, we have

$$\{ a, b..c \} := \{ a + n \cdot s \mid n \in \mathbb{N} \wedge a + n \cdot s \geq c \}.$$

For example, we have

$$\{ 10, 8 .. 1 \} = \{ 10, 8, 6, 4, 2 \}.$$

2.3.1 Operators on Sets

The most basic operators on sets are as follows:

1. “+” is used to compute the union of two sets.

In mathematics, the union of two sets s_1 and s_2 is written as $s_1 \cup s_2$. It is defined as

$$s_1 \cup s_2 := \{ x \mid x \in s_1 \vee x \in s_2 \}.$$

2. “*” computes the intersection of two sets. In mathematics, the intersection of two sets s_1 and s_2 is written as $s_1 \cap s_2$. It is defined as

$$s_1 \cap s_2 := \{x \mid x \in s_1 \wedge x \in s_2\}.$$

3. “-” computes differences of sets.

In mathematics, the difference of two sets s_1 and s_2 is written as $s_1 \setminus s_2$. It is defined as

$$s_1 \setminus s_2 := \{x \in s_1 \mid x \notin s_2\}.$$

4. “%” computes the *symmetric difference* of sets.

In mathematics, the symmetric difference of two sets s_1 and s_2 is written as $s_1 \triangle s_2$. It is defined as

$$s_1 \triangle s_2 := (s_1 \setminus s_2) \cup (s_2 \setminus s_1).$$

Therefore, the commands

```
s1 := { 1, 2 }; s2 := { 2, 3 };
print("s1 + s2 = $s1 + s2$");
print("s1 - s2 = $s1 - s2$");
print("s1 * s2 = $s1 * s2$");
print("s1 % s2 = $s1 % s2$");
```

will produce the following results:

```
s1 + s2 = {1, 2, 3}
s1 - s2 = {1}
s1 * s2 = {2}
s1 % s2 = {1, 3}
```

Also note the use of *string interpolation* in the last example. This will be discussed in more detail in the section on strings.

For all of the above operators there is a variant of the assignment operator that includes the operator. For example, we can write

```
s += {x};
```

to add the element x to the set s , while the command

```
s -= {x};
```

removes the element x from the set s .

There is no operator to compute the Cartesian product $s \times t$ of two given sets s and t . In mathematics, this product is defined as

$$s \times t := \{\langle x, y \rangle \mid x \in s \wedge y \in t\}.$$

However, the Cartesian product is readily calculated by the expression

```
{ [x, y] : x in s, y in t }.
```

In addition to the basic operators, SETLX provides a number of more elaborate operators for sets. One of these operators is the cardinality operator “#”, which computes the number of elements of a given set. The cardinality operator is used as a unary prefix operator, for example we can write:

```
# {5,7,13};
```

Of course, in this example the result will be 3. We can sum the elements of a set using the prefix operator “+”. In order to compute the sum

$$\sum_{i=1}^{6^2} i,$$

we can use the command

```
+ / { 1..6**2 };
```

For arguments that are numbers, “**” is the exponentiation operator. Therefore, the expression given above first computes the set of all numbers from 1 up to the number 36 and then returns the sum of all these numbers. There is also a binary version of the operator “+ /” that is used as an infix operator. For a number x and a set s , the expression

```
x + / s
```

will insert x into the set s if the set s is empty. After that, it returns the sum of all elements in the resulting set. This is useful as the expression “+ / {}” is undefined. So if we want to compute the sum of all numbers in s but we are not sure whether s might be empty, we can always use the expression

```
0 + / s
```

since inserting 0 into the empty set guarantees that the result is 0 in case the set s is empty.

There is a similar operator for multiplying the elements of a set: It is the operator “*/”. For example, in order to compute the factorial³ $n! = 1 \cdot 2 \cdot 3 \cdot \dots \cdot (n-1) \cdot n$ we can use the expression

```
*/ {1..n}.
```

Again, there is also a binary version of this operator. Since the number 1 is the neutral element for multiplication, the first argument to this operator will most often be one. Therefore, the expression

```
1 */ s
```

multiplies all the numbers from a the set s and this expression will return 1 if the set s happens to be empty.

2.3.2 Set Comprehensions

We can use *set comprehension* to build sets. For example, the command

```
{ a * b : a in { 1 .. 3 }, b in { 1 .. 3 } };
```

computes the set of all products $a * b$ where both a and b run from 1 to 3. The command will therefore compute the set

```
{1, 2, 3, 4, 6, 9}.
```

In general, a *set comprehension expression* has the form

```
{ expr : x1 in s1, ..., xn in sn | cond }.
```

Here, *expr* is some expression containing the variables x_1, \dots, x_n , while s_1, \dots, s_n are either sets or lists (to be discussed later), and *cond* is an expression returning a Boolean value. The set comprehension expression given above evaluates *expr* for all possible combinations of values $x_1 \in s_1, \dots, x_n \in s_n$ such that *cond* is true and it will add the corresponding value of *expr* into the resulting set. The expression *cond* is optional. If it is missing, it is implicitly taken to be always true.

We are now ready to demonstrate some of the power that comes with sets. The following two statements compute the set of all prime numbers smaller than 100:

```
s := {2..100};
s - { p * q : p in s, q in s };
```

When used on sets, the operator “-” computes the difference set. The expression given above

³ The factorial operator “!” is a builtin postfix operator. Computing the factorial of a number using this operator is more efficient than first building a set and then using the operator “*/”.

computes the set of prime numbers smaller than 100 correctly, as a prime number is any number bigger than 1 that is not a proper product. Therefore, if we subtract the set of all proper products from the set of numbers, we get the set of prime numbers. Obviously, this is not an efficient way to calculate primes, but efficiency is not the point of this example. It rather shows that SETLX is well suited to execute a mathematical definition as it is.

There is a short form available for set comprehension: For a set s and a Boolean expression $cond$, the expression

```
{ x in s | cond };
```

will compute the subset of all those elements of the set s that satisfy the specified Boolean condition $cond$. For example, the expression

```
{ p in {1..n} | n % p == 0 }
```

computes the set of all those numbers p that divide n without leaving a remainder.

2.3.3 Miscellaneous Set Functions

In addition to the operators provided for sets, SETLX has a number of functions targeted at sets. The first of these functions is **arb**. For a set s , an expression of the form

```
arb(s)
```

yields some element of the set s . The element returned is not specified, so the expression

```
arb({1,2,3});
```

might yield either 1, 2, or 3. The function **from** has a similar effect, so

```
from(s)
```

also returns some element of the set s . In addition, this element is removed from the set. Therefore, if the set s is defined via

```
s := {1,2,3};
```

and we execute the statement

```
x := from(s);
```

then afterwards the set s will no longer contain the element x . For a set s , the expression

```
first(s)
```

returns the *first* element of the set s , while

```
last(s)
```

yields the *last* element. Internally, sets are represented as ordered binary trees. As long as the sets contain only rational numbers, the ordering of these numbers is the usual ordering on numbers. If the sets contain sets or lists, these sets are themselves compared lexicographically. Things get more complicated if a set is heterogenous and contains both number and sets. Programs should not rely on the ordering implemented in these cases.

For a given set s , the expression

```
rnd(s)
```

yields a random element of the set s . In contrast to the expression “**arb**(s)”, this expression will in general return different results on different invocations. The function **rnd** also works for lists, so for a list l the expression **rnd**(l) returns a random element of l .

The function **rnd**() is also supported for integer numbers, for example the expression

```
rnd(5)
```

computes a random natural number less or equal than 5. Essentially it is equal to `rnd({1..5})`;

2.4 Lists

Besides sets, SETLX also supports lists. Syntactically, the main difference is that the braces “{” and “}” are substituted with the square brackets “[” and “]”. Semantically, a list is an ordered collection of elements that can contain the same element multiple times. For example

`[1,4,7,2,4,7]`

is a typical list. The easiest way to construct a list is via the range operator “..”: For example, the command

`[1..10]`

computes the result

`[1, 2, 3, 4, 5, 6, 7, 8, 9, 10].`

The prefix operators “+” and “*” work for lists also. For example,

`+ [1..100]`

yields the result 5050.

List comprehension works similar to set comprehension, the expression

`[expr : x1 in s1, ..., xn in sn | cond]`

picks up all tuples

$\langle x_1, \dots, x_n \rangle \in s_1 \times \dots \times s_n$

such that *cond* is true and evaluates *expr* for the corresponding values of the variables x_1, \dots, x_n . The value of *expr* is then inserted into the resulting list. Here, the s_i denote either sets or lists. For example, the following expression computes all primes up to 100:

`[p : p in [2..100] | { x : x in {1..p} | p % x == 0 } == {1, p}].`

Here, the idea is that a number p is prime if the set of its divisors, which is the set of all numbers x such that $p \% x = 0$, only contains the number 1 and the number p .

There are two infix operators for lists: The first is the operator “+” which concatenates its arguments. For example

`[1 .. 3] + [5 .. 10];`

yields

`[1, 2, 3, 5, 6, 7, 8, 9, 10].`

The second infix operator is “*”. While the first argument of this operator is a list, the second needs to be an integer. An expression of the form

`1 * n`

concatenates *n* copies of the list 1. Therefore, the expression

`[1, 2, 3] * 3`

yields the result

`[1, 2, 3, 1, 2, 3, 1, 2, 3].`

The cardinality operator “#” works for lists the same way it works for sets: It returns the number of elements. For example

`# [7, 4, 5];`

yields the result 3.

As the elements of a list are ordered, it is possible to extract an element from a list with respect to its position. In general, for a list l and a positive natural number n , the expression

$$l[n]$$

selects the n -th element of l . Here, counting starts with 1, so $l[1]$ is the first element of the list l . For example, after the assignment

$$l := [99, 88, 44];$$

the expression

$$l[2]$$

yields the result 88. The right hand side of an extraction expression can also be a *slicing operator*: The expression

$$l[a..b]$$

extracts the sublist of l that starts at index a and ends at index b . For example, after defining l via the assignment

$$l := [1..100];$$

the expression

$$l[5..10]$$

yields the sublist $[5, 6, 7, 8, 9, 10]$. There are two variants of the slicing operator. The expression

$$l[a..]$$

returns the sublist of l that starts with the element at index a , and the expression

$$l[..b]$$

computes the sublist of l that starts at the first element and includes all the elements up to the element at index b .

Lists can also be used on the left hand side of an assignment. The statement

$$[x, y] := [1, 2];$$

sets the variable x to 1 and y to 2. This feature can be used to swap the values of variables: the assignment

$$[y, x] := [x, y];$$

swaps the values of x and y .

2.5 Pairs, Relations, and Functions

In SETL, a pair of the form $\langle x, y \rangle$ is represented as the list $[x, y]$. A set of pairs can be regarded as a binary relation and the notion of a relation is a generalization of the notion of a function. If r is a binary relation, we define the *domain* and *range* of r as the set containing the first and second component of the pairs, that is we have

$$\text{domain}(r) = \{ x : [x, y] \text{ in } r \} \quad \text{and} \quad \text{range}(r) = \{ y : [x, y] \text{ in } r \}.$$

Furthermore, if r is a binary relation such that for all pairs

$$[x, y_1] \in r \quad \text{and} \quad [x, y_2] \in r$$

we have that $y_1 = y_2$, then r is called a *map* and represents a function. Therefore, a map is a set of **[key, value]** pairs such that the keys are unique. If a binary relation r is a map, SETLX

permits us to use the relation as a function: If r is a map and $x \in \text{domain}(r)$, then $r[x]$ denotes the unique element y such that $\langle x, y \rangle \in r$:

$$r[x] := \begin{cases} y & \text{if the set } \{y \mid [x, y] \in r\} \text{ contains exactly one element } y; \\ \Omega & \text{otherwise.} \end{cases}$$

The program `function.stlx` shown below in figure 2.4 shows how maps can be used as functions in SETLX.

```

1  r := { [n, n*n] : n in {1..10} };
2  print( "r[3]      = $r[3]$" );
3  print( "domain(r) = $domain(r) $" );
4  print( "range(r)  = $range(r) $" );

```

Figure 2.4: Binary relations as functions.

The program computes the map r that represents the function $x \mapsto x * x$ on the set $\{n \in \mathbb{N} \mid 1 \leq n \wedge n \leq 10\}$. In line 3, the relation is evaluated at $x = 3$. This is done using square brackets. Finally, $\text{domain}(r)$ and $\text{range}(r)$ are computed. We get the following result:

```

r[3]      = 9
domain(r) = {1, 2, 3, 4, 5, 6, 7, 8, 9, 10}
range(r)  = {1, 4, 9, 16, 25, 36, 49, 64, 81, 100}

```

It is a natural question to ask what happens if r is a binary relation and we try to evaluate the expression $r(x)$ but the set $\{y : [x, y] \text{ in } r\}$ is either empty or contains more than one element. The program shown in figure 2.5 on page 17 computes the answer to this question.

```

1  r := { [1, 1], [1, 4], [3, 3] };
2  print( "r[1] = ", r[1] );
3  print( "r[2] = ", r[2] );
4  print( "{ r[1], r[2] } = ", { r[1], r[2] } );
5  print( "r{1} = ", r{1} );
6  print( "r{2} = ", r{2} );

```

Figure 2.5: A binary relation that is not a map.

If the set $\{y : [x, y] \text{ in } r\}$ is either empty or has more than one element, then the expression $r[x]$ is undefined. In mathematics, an undefined value is sometimes denoted as Ω . In SETLX, the undefined value is printed as “om”. If we try to add the undefined value to a set m , then m is not changed. Therefore, line 5 of the program just prints the empty set, as both $r[1]$ and $r[2]$ are undefined.

We can use the notation $r\{x\}$ instead of $r[x]$ to avoid undefined values. For a binary relation r and an object x , the expression $r\{x\}$ is defined as follows:

$$r\{x\} := \{y : [x, y] \text{ in } r\}.$$

Therefore, the program from Figure 2.5 yields the following results:

```

r[1] = om
r[2] = om
{ r[1], r[2] } = {}
r{1} = {1, 4}
r{2} = {}

```

2.6 Strings

Any sequence of characters enclosed in double quotes is considered a string. Strings can be concatenated using the infix operator “+”, so

```
"abc" + "xyz"
```

yields the string "abcxyz" as a result. In order to concatenate multiple instances of the same string, we can use the infix operator “*”. For a string *s* and a natural number *n*, the expression

```
s * n
```

returns a string that consists of *n* copies of the string *s*. For example, the expression

```
"abc" * 3
```

yields the result

```
"abccabccabc".
```

This also works when the position of the string and the number are exchanged. Therefore,

```
3 * "abc"
```

also yields "abccabccabc".

In order to extract the *i*-th character of the string *s*, we can use the expression

```
s[i].
```

The slicing operators work similar to lists, for example, if *s* has the value "abcdef", then

```
s[2..5];
```

yields the result "bcde", while

```
s[2..];
```

yields "bcdef" and

```
s[..5];
```

gives "abcde".

SETLX provides *string interpolation*: If a string contains a substring enclosed in “\$”-symbols, then SETLX parses this substring as an expression and substitutes the result into the string. For example, if the variable *n* has the value 6, the command

```
print("$n$! = $n!$");
```

will print

```
6! = 720.
```

In order to insert a literal “\$”-symbol into a string, the “\$”-symbol has to be escaped with a backslash. For example, the command

```
print("a single \$-symbol");
```

prints the text

```
a single $-symbol.
```

String interpolation is turned off by the prefix operator “@”: The command

```
print(@"$n$! = $n!$");
```

prints

```
$n$! = $n!$.
```

2.7 Terms

SETLX provides first order terms similar to the terms available in the language *Prolog*. Terms are built from *functors* and *arguments*. To distinguish functors from function symbols, a functor will start with a capital letter. For example, the expression

`F(1,"x")`

is a term with functor `F` and two arguments. Here, the functor is just a name, it is not a function that can be evaluated. To demonstrate the usefulness of terms, consider implementing *ordered binary trees* in SETLX. There are two types of ordered binary tree:

1. The empty tree represents the empty set. We use the functor `Nil` to represent an empty binary tree, so the term

`Nil()`

codes the empty tree.

2. A non-empty binary tree has three components:

- (a) The *root* node,
- (b) the left subtree, and
- (c) the right subtree.

The root node stores one element k and all elements in the left subtree l have to be less than k , while all elements in the right subtree are bigger than k . Therefore, a non-empty binary tree can be represented as the term

`Node(k,l,r),`

where k is the element stored at the root, l is the left subtree and r is the right subtree.

For example, the term

`Node(2,Node(1,Nil(),Nil()), Node(3,Nil(), Nil()))`

is a typical binary tree. At the root, this tree stores the element 2, the left subtree stores the element 1 and the right subtree stores the element 3.

There are two functions to decompose a term and there is one function that constructs a term.

1. If t is a term, then the expression

`fct(t)`

returns the functor of the term t as a string. For example, the expression

`fct(Node(3,Nil(),Nil()))`

yields the result `"Node"`.

2. If t is a term, then the expression

`args(t)`

returns the list of arguments of the term t . For example, the expression

`args(Node(3,Nil(),Nil()))`

yields the result

`[3, Nil(), Nil()]`.

3. The function `makeTerm(f,l)` constructs a term t such that

`fct(t) = f` and `args(t) = l`

holds. For example,

```
makeTerm("Node", [ makeTerm("Nil",[]), makeTerm("Nil",[]) ])
```

constructs the term

```
Node(3,Nil(),Nil()).
```

Of course, this term can also be given directly as an expression. For example, the statement

```
a := Node(3,Nil(),Nil());
```

assigns the term `Node(3,Nil(),Nil())` to the variable `a`.

```

1  insert := procedure(m, k1) {
2      switch {
3          case fct(m) == "Nil" :
4              return Node(k1, Nil(), Nil());
5          case fct(m) == "Node":
6              [ k2, l, r ] := args(m);
7              if (k1 == k2) {
8                  return Node(k1, l, r);
9              } else if (compare(k1, k2) < 0) {
10                 return Node(k2, insert(l, k1), r);
11             } else {
12                 return Node(k2, l, insert(r, k1));
13             }
14         }
15     };

```

Figure 2.6: Inserting an element into a binary tree.

Figure 2.6 shows how terms can be used to implement binary trees. In this example, we define a function with the name `insert`. This function takes two arguments. The first argument `m` is supposed to be a term representing an ordered binary tree. The second argument `k1` denotes the number that is to be inserted into the binary tree `m`. The implementation needs to distinguish two cases:

1. If the binary tree `m` is empty, then the function returns the tree

```
Node(k1, Nil(), Nil()).
```

This is a binary tree containing the number `k1` at its root. In order to check whether `m` is indeed empty we use the functor of the term `m`. We test in line 3 whether this functor is `"Nil"`.

2. If the binary tree `m` is nonempty, the functor of `m` is `"Node"`. In this case, we need to extract the arguments of this functor, which is done in line 6: The first argument `k2` is the number stored at the root, while the arguments `l` and `r` correspond to the left and the right subtree of `m` respectively.

This example uses a number of features of SETLX that have not been introduced. The discussion of the `switch` construct will be given in the next chapter.

Sometimes, the fact that a functor has to be written in uppercase is inconvenient. For these cases, there is an escape mechanism: If the functor of a term is preceded by the symbol `"@"`, then it may start with a lowercase letter. For example, if we want to have terms representing algebraic expressions involving transcendental functions, we could use the same function names that are also

used in SETLX to denote the corresponding transcendental functions. For example, we can write

```
t := @exp(@log(@x));
```

to represent the term

```
exp(log(x)).
```

The present discussion of terms is not complete and will be continued in the next chapter when we discuss matching.

Chapter 3

Writing Statements in SetlX

This section discusses the various possibilities to write statements and the features offered by SETLX to steer the control flow in a program. SETLX supports the following statements to control the flow of execution:

1. branching statements like `if-then-else`, `switch`, and `match`,
2. the looping statements `for` and `while` together with `break` and `continue`,
3. `try-catch` statements to deal with exceptions,
4. the `backtrack` statement to support backtracking.

3.1 Assignment Statements

The most basic command is the assignment. In contrast to the programming languages C and Java, SETLX uses the operator “:=” for assignments. For example, the statement

```
x := 2/3;
```

binds the variable `x` to the fraction $\frac{2}{3}$. SETLX supports simultaneous assignments to multiple variables via lists. For example, the statement

```
[x, y, z] := [1, 2, 3];
```

simultaneously binds the variables `x` to 1, `y` to 2 and `z` to 3. This feature can be used to swap the values of two variables: The statement

```
[x, y] := [y, x];
```

swaps the values of `x` and `y`. If we do not need to assign all the values of a list, we can use the underscore “_” as an anonymous variable. For example,

```
[x, _, z] := [1, 2, 3];
```

assigns the number 1 to the variable `x` and the variable `z` is set to 3.

The assignment operator can be combined with any of the operators “+”, “-”, “*”, “/”, “%”, and “\”. For example, the statement

```
x += 1;
```

increments the value of the variable `x` by one, while the statement

```
x *= 2;
```

doubles the value of `x`.

Finally, assignment statements can be chained. For example, the statement

```
a := b := 3;
```

assigns the value 3 to both **a** and **b**.

3.2 Functions

The code shown in figure 3.1 on page 23 shows a simple program to compute prime numbers. It defines two functions. The function **factors** takes a natural number p as its first argument and computes the set of all factors of p . Here, a number f is a factor of p iff dividing p by f leaves no remainder, that is $p \% f = 0$. The second function **primes** takes a natural number n as an argument and computes the set of all those prime numbers that are less or equal to n .

The idea is that a number p is prime iff the set of all factors of p contains just 1 and p . Note that, as SETLX is a functional language, the functions that are defined by the keyword **procedure** are assigned to variables. These functions can be used like any other values.

```

1  factors := procedure(p) {
2      return { f in { 1 .. p } | p % f == 0 };
3  };
4  primes := procedure(n) {
5      return { p in { 2 .. n } | factors(p) == { 1, p } };
6  };
7  print(primes(100));

```

Figure 3.1: A naive program to compute primes.

A simplified grammar rule for the definition of a function can be given as follows:

$$fctDef \rightarrow \text{VAR} \text{ “:=” “procedure” “(” paramList “)” “{” block “}” “;”}$$

The meaning of the symbols used in this grammar rule are as follows:

1. **VAR** identifies a variable. This variable is bound to the definition of the function. Note that in SETLX the name of a variable has to start with a lower case letter.
2. *paramList* is a list of the formal parameters of the function. In EBNF-notation the grammar rule for *paramList* is given as

$$paramList \rightarrow (paramSpec \text{ “,” } paramSpec)^*$$

Therefore, a *paramList* is a possibly empty list of parameter specifications that are separated by a comma “,”. A parameter specification is either just a variable or it is a variable preceded by the token “**rw**”:

$$paramSpec \rightarrow (ID \mid \text{“rw” } ID).$$

If the parameter is preceded by the keyword **rw**, then this parameter is a *read-write parameter*, which means that the function can change the value of the variable given as argument. Parameters not specified as read-write parameters have a strict *call by value semantics*, and therefore changes to those parameters will not be visible outside the function.

3. *block* is a sequence of statements.

Note that the definition of a function has to be terminated by the symbol “;”.

There is a variant syntax for defining function which is appropriate if the definition of the function is just a single expression. For example, the function mapping x to the square $x * x$ can be defined as

```
f := x |-> x * x;
```


A definition of this form is called a *lambda definition*. If the function takes more than one arguments, the argument list has to be enclosed in square brackets. For example, the function `hyp` that computes the hypotenuse of a rectangular triangle can be defined as follows:

```
hyp := [x, y] |-> sqrt(x*x + y*y);
```

The syntax for a lambda definition is given by the following grammar rule:

$$fctDef \rightarrow ID \text{ “:=” } lambdaParams \text{ “|->” } expr$$

Here, *lambdaParams* is either just a single parameter or a list of parameters, where the parameters are enclosed in square brackets and are separated by commas, while *expr* denotes an expression.

Lambda definitions are handy if we don’t bother to give a name to a function. For example, the code in figure 3.2 defines a function `map`. This function takes two arguments: The first argument *l* is a list and the second argument *f* is a function that is to be applied to all arguments of this list. In line 4, the function `map` is called with a function that squares its argument. Therefore, the assignment in line 4 computes the list of the first 10 square numbers. Note that we did not had to name the function that did the squaring. Instead, we have used a lambda definition.

```

1  map := procedure(l, f) {
2      return [ f(x) : x in l ];
3  };
4  t := map([1 .. 10], x |-> x * x);

```

Figure 3.2: An example of a lambda definition in use.

Of course, it is much easier to build the list of the first 10 squares using the statement

```
t := [x*x : x in [1..10]];
```

3.2.1 Memoization

The Fibonacci function $fib : \mathbb{N} \rightarrow \mathbb{N}$ is defined recursively by the following equations:

$$fib(0) = 0, \quad fib(1) = 1, \quad \text{and} \quad fib(n+2) = fib(n+1) + fib(n).$$

These equations are readily implemented as shown in Figure 3.3. However, this implementation has a performance problem which can be easily seen when tracing the computation of $fib(4)$.

```

1  fibonacci := procedure(n) {
2      if (n in [0,1]) {
3          return n;
4      }
5      return fibonacci(n-1) + fibonacci(n-2);
6  };

```

Figure 3.3: A naive implementation of the Fibonacci function.

```

1  fibonacci := procedure(n) {
2      if (n in [0,1]) {
3          result := n;
4      } else {
5          result := fibonacci(n-1) + fibonacci(n-2);
6      }
7      print("fibonacci($n$) = $result$");
8      return result;
9  };

```

Figure 3.4: Tracing the computation of the Fibonacci function.

In order to trace the computation, we change the program as shown in Figure 3.4. If we evaluate the expression `fibonacci(4)`, we get the output shown in Figure 3.5. This output shows that the expression `fibonacci(2)` is evaluated twice. The reason is that the value of `fibonacci(2)` is needed in the equation

$$\text{fibonacci}(4) := \text{fibonacci}(3) + \text{fibonacci}(2)$$

to compute `fibonacci(4)`, but then in order to compute `fibonacci(3)`, we have to compute `fibonacci(2)` again. The trace also shows that the problem gets aggravated the longer the computation runs. For example, the value of `fibonacci(1)` has to be computed three times.

```

=> fibonacci(4);

fibonacci(1) = 1
fibonacci(0) = 0
fibonacci(2) = 1
fibonacci(1) = 1
fibonacci(3) = 2
fibonacci(1) = 1
fibonacci(0) = 0
fibonacci(2) = 1
fibonacci(4) = 3
~< Result: 3 >~

=>

```

Figure 3.5: Output of evaluating the expression `fibonacci(4)`.

In order to have a more efficient computation, it is necessary to memorize the values of the function `fibonacci` once they are computed. Fortunately, SETLX offers *cached functions*. If a function f is declared as a cached function, then every time the function f is evaluated for an argument x , the computed value $f(x)$ is memorized and stored in a table. The next time the function f is used to compute $f(x)$, the interpreter first checks whether the value of $f(x)$ has already been computed. In this case, instead of computing $f(x)$ again, the function returns the value stored in the table. Figure 3.6 shows an implementation of the Fibonacci function as a cached function. If we compare the program in Figure 3.6 with our first attempt shown in Figure 3.3, then we see that the only difference is that instead of the keyword “`procedure`” we have used the keyword “`cachedProcedure`” instead.

Warning: A function should only be declared as a `cachedProcedure` if it is guaranteed to always produce the same result when called with the same argument. Therefore, a function should not be declared as a `cachedProcedure` if it does one of the following things:

```

1  fibonacci := cachedProcedure(n) {
2      if (n in [0,1]) {
3          return n;
4      }
5      return fibonacci(n-1) + fibonacci(n-2);
6  };

```

Figure 3.6: A cached implementation of the Fibonacci function.

1. The function reads global variables that might change.

A variable `x` can be declared to be a global variable by writing

```
var x;
```

In that case, the variable is visible in the scope of all functions.

2. The function makes use of random numbers.
3. The function reads input either from a file or from the command line.

To further support cached procedures, SETLX provides the function `cacheStats`, which is called with a single argument that must be a cached function. For example, if we define the function `fibonacci` as shown in Figure 3.6 and evaluate the expression `fibonacci(100)`, then the expression `cacheStats(fib)` gives the following result:

```
~< Result:  ["cache hits", 98], ["cached items", 101] >~
```

This tells us that the cache contains 101 different argument/value pairs, as the cache now stores the values

`fibonacci(n)` for all $n \in \{0, \dots, 100\}$.

Furthermore, we see that 98 of these 101 argument/value pairs have been used in order to compute the values of `fibonacci` for different arguments. The reason is that the values for the last three arguments 99, 100, and 101 have not yet been used for the computation of different values, but all other arguments have been used at least once for computing another value.

In order to prevent memory leaks, SETLX provides the function `clearCache`. This function is invoked with one argument which must be a cached function. Writing

```
clearCache(f)
```

clears the cache for the function f , that is all argument/value pairs stored for f will be removed from the cache.

3.3 Branching Statements

Like most modern languages, SETLX supports **if-then-else** statements and **switch** statements. A generalization of **switch** statements, the so called **match** statements, are also supported. We begin our discussion with **if-then-else** statements.

3.3.1 if-then-else Statements

In order to support branching, SETLX supports **if-then-else** statements. The syntax is similar to the corresponding syntax in the programming language C. However, braces are required. For example, figure 3.7 on page 27 shows a recursive function that computes the binary representation of a natural number.

As in the programming languages C and Java, the **else** clause is optional.

```

1  toBin := procedure(n) {
2      if (n < 2) {
3          return str(n);
4      } else {
5          r := n % 2;
6          n := floor(n / 2);
7          return toBin(n) + toBin(r);
8      }
9  };

```

Figure 3.7: A function to compute the binary representation of a natural number.

3.3.2 switch Statements

Figure 3.8 shows a function that takes a list of length 3. The function sorts the resulting list. In effect, the function `sort3` implements a decision tree. This example shows how `if-else` statements can be cascaded.

```

1  sort3 := procedure(l) {
2      [ x, y, z ] := l;
3      if (x <= y) {
4          if (y <= z) {
5              return [ x, y, z ];
6          } else if (x <= z) {
7              return [ x, z, y ];
8          } else {
9              return [ z, x, y ];
10         }
11     } else if (z <= y) {
12         return [z, y, x];
13     } else if (x <= z) {
14         return [ y, x, z ];
15     } else {
16         return [ y, z, x ];
17     }
18 };

```

Figure 3.8: A function to sort a list of three elements.

Figure 3.9 on page 28 shows how to sort a list of length 3 using a switch statement. Although this implementation is easier to understand, it is less efficient than the previous version. The reason is that some of the tests are redundant. This is most obvious for the last case in line 9 since at the time when control arrives in line 9 it is already known that z must be less or equal than y and that, furthermore, y must be less or equal than x , since all other cases are already covered.

The grammar rule for `switch` statements is as follows:

$$stmnt \rightarrow \text{"switch"} \text{"{" } caseList \text{"}"}$$

where the syntactical variable *caseList* is defined via the rule:

$$caseList \rightarrow (\text{"case"} \text{ boolExpr } \text{":" } block) * (\text{"default"} \text{":" } block)?$$

Here, *boolExpr* is a boolean expression and *block* is a list of statements.

```

1  sort3 := procedure(l) {
2      [ x, y, z ] := l;
3      switch {
4          case x <= y && y <= z: return [ x, y, z ];
5          case x <= z && z <= y: return [ x, z, y ];
6          case y <= x && x <= z: return [ y, x, z ];
7          case y <= z && z <= x: return [ y, z, x ];
8          case z <= x && x <= y: return [ z, x, y ];
9          case z <= y && y <= x: return [ z, y, x ];
10         default: print("Impossible error occurred!");
11     }
12 };

```

Figure 3.9: Sorting a list of 3 elements using a **switch** statement.

In contrast to the programming languages **C** and *Java*, the **switch** statement in SETLX doesn't have a fall through. Therefore, we don't need a **break** statement in the block of statements following a case condition. There are two other important distinction between the **switch** statement in *Java* and the **switch** statement in SETLX:

1. The keyword **switch** is not followed by a value.
2. The conditions following the keyword **case** have to be Boolean values.

There is a another type of switching statements that is much more powerful than the **switch** statement. This is called *matching* and will be discussed in the next section.

3.4 Matching

One of the most powerful branching construct is *matching*. Although the syntax for the matching statement is always the same, there are really four different types of matching for each of the types that matching has been implemented. Matching has been implemented for strings, list, sets, and terms. We discuss *string matching* first.

3.4.1 String Matching

Many algorithms that deal with a given string s have to deal with two cases: Either the string s is empty or it is nonempty and has to be split into its first character c and the remaning characters r , that is we have

$$s = c + r \quad \text{where } c = s[1] \text{ and } r = s[2..].$$

In order to facilitate algorithms that have to make this kind of case distinction, SETLX provides the **match** statement. Consider the function¹ **reverse** shown in Figure 3.10. This function reverses its input argument, so the expression

```
reverse("abc")
```

yields the result "cba". In order to reverse a string s , the function has to deal with two cases:

1. If the string s is empty. In this case, we can just return the string s as it is. This case is dealt with in line 3. There, we have used the pattern `[]` to match the empty string. Instead, we

¹ There is also a predefined version of the function **reverse** which does exactly the same thing as the function in Figure 3.10. However, once we define the function **reverse** as in Figure 3.10, the predefined function gets overwritten and is no longer accessible.

could have used the empty string itself. Using the pattern “[]” will prove beneficial when dealing with lists because it turns out that the function `reverse` as given in Figure 3.10 can also be used to reverse a list.

2. If the string s is not empty, then it can be split up into a first character c and the remaining characters r . In this case, we reverse the string r and append the character c to the end of this string. This case is dealt with in line 4.

The `match` statement in function `reverse` has a default case in line 5 to deal with those cases where s is not a string.

```

1  reverse := procedure(s) {
2      match (s) {
3          case [] : return s;
4          case [c|r]: return reverse(r) + c;
5          default : abort("type error in reverse($s$)");
6      }
7  };

```

Figure 3.10: A function that reverses a string.

The last example shows that the syntax for the `match` statement is similar to the syntax for the `switch` statement. The main difference is that the keyword is now “`match`” instead of “`switch`” and that the cases no longer contain Boolean values but instead contain *patterns* that can be used to check whether a string has a given form and to extract the corresponding components. Basically, the function `reverse` given above is interpreted as if it had been written in the way shown in Figure 3.11 on page 29. This example shows that the use of the `match` statement can make programs more compact while increasing their legibility.

```

1  reverse := procedure(s) {
2      if (s == "") {
3          return s;
4      } else if (isString(s)) {
5          c := s[1];
6          r := s[2..];
7          return reverse(r) + c;
8      } else {
9          abort("type error in reverse($s$)");
10     }
11 };

```

Figure 3.11: A function to reverse a string that does not use matching.

To explore string matching further, consider a function `reversePairs` that interchanges all pairs of characters, so for example we have

`reversePairs("abcd") = "badc" and reversePairs("abcde") = "badce".`

This function can be implemented as shown in Figure 3.12 on page 30. Notice that we can match the empty string with the pattern “[]” in line 3. The pattern “[c]” matches a string consisting of one character c . In line 5 the pattern `[a,b|r]` extracts the first two characters from the string s and binds them to the variables a and b . The rest of the string is then bound to r .

```

1  reversePairs := procedure(s) {
2      match (s) {
3          case []      : return s;
4          case [c]     : return c;
5          case [a,b|r]: return b + a + reversePairs(r);
6
7      }
8  };

```

Figure 3.12: A function to exchange pairs of characters.

3.4.2 List Matching

As strings can be seen as lists of characters, the matching of lists is very similar to the matching of strings. The function `reverse` shown in Figure 3.10 can also be used to reverse a list. If the argument s of `reverse` is a list instead of a string, we match the empty list with the pattern “`[]`”, while the pattern “`[c|r]`” matches a non-empty list: The variable c matches the first element of the list, while the variable r matches the remaining elements.

3.4.3 Set Matching

As sets are quite similar to lists, the matching of sets is closely related to the matching of lists. Figure 3.13 shows the function `setSort` that takes a set of numbers as its argument and returns a sorted list containing the numbers appearing in the set. In the `match` statement, we match the empty set with the pattern “`{}`”, while the pattern “`{x|r}`” matches a non-empty set: The variable x matches the first element of the set, while the variable r matches the set of all the remaining elements.

```

1  setSort := procedure(s) {
2      match (s) {
3          case {}      : return [];
4          case {x|r}: return [x] + setSort(r);
5      }
6  };

```

Figure 3.13: A function to sort a set of numbers.

Of course, sorting a set into a list is trivial, as a set in SETLX is represented by an ordered binary tree and therefore is already sorted. For this reason, we could also transform a set s into a sorted list by using the expression

`[] + s.`

In general, for a list l and a set s the expression $l + s$ creates a new list containing all elements of l . Then, the elements of s are appended to this list. The expression $s + l$ creates a new set containing the elements of s . Then, the elements of l are inserted into this set.

3.4.4 Term Matching

The most elaborate form of matching is the matching of terms. This kind of matching is similar to the kind of matching provided in the programming languages *Prolog* and *ML*. Figure 3.14 shows an implementation of the function `insert` that uses term matching instead of the functions “`fct`”

and “args” that had been used in the previous implementation shown in Figure 2.6 on page 20. In line 3 of Figure 3.14, the `case` statement checks whether m is the empty tree. This is more straightforward than testing that the functor of m is “Nil”, as it is done in line 3 of Figure 2.6. However, the real benefit of matching shows in line 5 of Figure 3.14 since the case statement in this line does not only check whether the functor of m is “Node” but also assigns the variables `k2`, `l`, and `r` to the respective subterms of m . Compare this with line 5 and line 6 of Figure 2.6 where we had to use a separate statement in line 6 to extract the arguments of m .

```

1  insert := procedure(m, k1) {
2      match (m) {
3          case Nil() :
4              return Node(k1, Nil(), Nil());
5          case Node(k2, l, r):
6              if (k1 == k2) {
7                  return Node(k1, l, r);
8              } else if (compare(k1, k2) < 0) {
9                  return Node(k2, insert(l, k1), r);
10             } else {
11                 return Node(k2, l, insert(r, k1));
12             }
13         default: abort("Error in insert($m$, $k1$, $v1$)");
14     }
15 };

```

Figure 3.14: Inserting an element into a binary tree using matching.

Let us discuss a more complex example of matching. The function `diff` shown in Figure 3.15 on page 32 is supposed to be called with two arguments:

1. The first argument t is a term that is interpreted as an arithmetic expression.
2. The second argument x is a string that is interpreted as the name of a variable.

The function `diff` interprets the term t as a mathematical function and differentiates this function with respect to the variable x . For example, in order to compute the derivative of the function

$$x \mapsto x^x$$

we can invoke `diff` as follows:

```
diff(parse("x ** x"), "x");
```

Here the function `parse` transforms the string “`x ** x`” into a term. The form of this term will be discussed in more detail later. For the moment, let us focus on the `match` statement in Figure 3.15. Consider line 3: If the term that is to be differentiated has the form `t1 + t2`, then both `t1` and `t2` have to be differentiated separately and the resulting terms have to be added. For a more interesting example, consider line 8. This line implements the product rule:

$$\frac{d}{dx}(t_1 \cdot t_2) = \frac{dt_1}{dx} \cdot t_2 + t_1 \cdot \frac{dt_2}{dx}.$$

Note how the pattern

```
t1 * t2
```

in line 7 extracts the two factors from a term that is a product. Further, note that in line 12 and line 16 we had to prefix the function symbols “`exp`” and “`ln`” with the character “`@`” in order to convert these function symbols into functors.

```

1  diff := procedure(t, x) {
2      match (t) {
3          case t1 + t2 :
4              return diff(t1, x) + diff(t2, x);
5          case t1 - t2 :
6              return diff(t1, x) - diff(t2, x);
7          case t1 * t2 :
8              return diff(t1, x) * t2 + t1 * diff(t2, x);
9          case t1 / t2 :
10             return ( diff(t1, x) * t2 - t1 * diff(t2, x) ) / t2 * t2;
11         case f ** g :
12             return diff( @exp(g * @ln(f)), x);
13         case ln(a) :
14             return diff(a, x) / a;
15         case exp(a) :
16             return diff(a, x) * @exp(a);
17         case ^variable(x) : // x is defined above as second argument
18             return 1;
19         case ^variable(y) : // y is undefined, matches any other variable
20             return 0;
21         case n | isNumber(n):
22             return 0;
23     }
24 };

```

Figure 3.15: A function to perform symbolic differentiation.

In order to understand this example in Figure 3.15 in more detail, we have to discuss how the function `parse` converts a string into a term. The function `parse` needs to represent all operator symbols and it also needs to represent variables. A variable of the form `"x"` is parsed as a term of the form

`^variable("x").`

This should explain the patterns used in line 19 and line 21 of Figure 3.15. In order to inspect the internal representation of a term, we use the function `"canonical"`. For example, the expression

`canonical(parse("x ** x"))`

yields the result

`^power(^variable("x"), ^variable("x")).`

This shows that the functor `"^power"` is the internal representation of the power operator `"**"`. The internal representation of `"+"` is `"^sum"`, `"-"` is represented as `"^difference"`, `"*"` is represented as `"^product"`, and `"/"` is represented as `"^quotient"`.

Note that the example makes extensive use of the fact that terms are *viral* when used with the arithmetic operators `"+"`, `"-"`, `"*"`, `"/"`, `"\"`, and `"%"`: If one operand of these operators is a term, the operator automatically yields a term. For example, the expression

`parse("x") + 2`

yields the term

`^sum(^variable("x"), 2).`

Note also that terms are not viral inside function symbols like `"exp"`. Therefore, this function

symbol has to be prefixed by the operator “@” to turn it into a functor.

Line 21 shows how a condition can be attached to a pattern: The pattern

```
case n:
```

would match anything. However, we want to match only numbers here. Therefore, we have used the pattern

```
case n | isNumber(n):
```

in order to ensure that `n` is indeed a number.

3.5 Loops

SETLX offers two different kinds of loops: `for` loops and `while` loops. The `while` loops are the most general loops. Therefore, we discuss them first.

3.5.1 while Loops

The syntax and semantics of `while` loops in SETLX is really the same as in the programming language `C`. To demonstrate a `while` loop, let us implement a function testing the *Ulam conjecture*: Define the function

$$f : \mathbb{N} \rightarrow \mathbb{N}$$

recursively as follows:

1. $f(n) := 1$ if $n \leq 1$,
2. $f(n) := \begin{cases} f(n/2) & \text{if } n \% 2 = 0; \\ f(3 \cdot n + 1) & \text{otherwise.} \end{cases}$

Then f is well-defined and, furthermore, $f(n) = 1$ for all $n \in \mathbb{N}$. Figure 3.16 shows an implementation of this function in SETLX.

```

1  f := procedure(n) {
2      if (n == 0) {
3          return 1;
4      }
5      while (n != 1) {
6          if (n % 2 == 0) {
7              n /= 2;
8          } else {
9              n := 3 * n + 1;
10         }
11     }
12     return n;
13 };

```

Figure 3.16: A program to test the Ulam conjecture.

The function f is implemented via a `while` loop. This loop runs as long as n is different from 1. The syntax of a `while` loop is given by the following grammar rule:

$$\textit{statement} \rightarrow \textit{“while” “(” boolExpr “)” “{” block “}”}.$$

Here, *boolExpr* is a Boolean expression returning either `true` or `false`. This condition is called

the *guard* of the **while** loop. *block* is a list of statements. Note that the block of statements always has to be enclosed in curly braces, even if it consists only of a single statement. The semantics of a **while** loop is the same as in C: The loop is executed as long as the guard is true. In order to abort an iteration prematurely, SETLX provides the command **continue**. This command aborts the current iteration of the loop and proceeds with the next iteration. In order to abort the loop itself, the command **break** can be used. Figure 3.17 shows a function that uses both a **break** statement and a **continue** statement. This function will print the number 1. Then, when *n* is incremented to 2, the **continue** statement in line 6 is executed so that the number 2 is not printed. In the next iteration of the loop, the number *n* is incremented to 3 and printed. In the final iteration of the loop, *n* is incremented to 4 and the **break** statement in line 9 terminates the loop.

```

1  testBreakAndContinue := procedure() {
2      n := 0;
3      while (n < 10) {
4          n := n + 1;
5          if (n == 2) {
6              continue;
7          }
8          if (n == 4) {
9              break;
10         }
11         print(n);
12     }
13 };

```

Figure 3.17: This function demonstrates the semantics of **break** and **continue**.

3.5.2 for Loops

In order to perform a list of commands a predefined number of times, a **for** loop should be used. Figure 3.18 on page 35 shows some SETLX code that prints a multiplication table. The output of this program is shown in Figure 3.19 on page 35.

In the program in Figure 3.18, the printing is done in the two nested loops that start in line 8. In line 8, the counting variable *i* iterates over all values from 1 to 10. Similarly, the counting variable *j* in line 9 iterates over the same values. The product *i* * *j* is computed in line 10 and printed without a newline using the function **nPrint**. The function **rightAdjust(*n*)** turns the number *n* into a string by padding the number with blanks from the left so that the resulting string always has a length of 4 characters.

The general syntax of a **for** loop is given by the following EBNF rule:

$$\textit{statement} \rightarrow \textit{"for"} \textit{"("} \textit{iterator} \textit{"("} \textit{","} \textit{iterator} \textit{")"} \textit{"("} \textit{"{"} \textit{block} \textit{"}"}$$

Here an iterator is either a *simple iterator* or a *tuple iterator*. A *simple iterator* has the form

$$x \textit{"in"} s.$$

Here *s* is either a set, a list, or a string and *x* is the name of a variable. This variable is bound to the elements of *s* in turn. For example, the statement

$$\textit{for} \textit{"(x in [1..10])"} \textit{"{"} \textit{print(x)} \textit{";"}$$

will print the numbers from 1 to 10. If *s* is a string, then the variable *x* iterates over the characters of *s*. For example, the statement

```

1  rightAdjust := procedure(n) {
2      switch {
3          case n < 10 : return "  " + n;
4          case n < 100: return " " + n;
5          default:    return " " + n;
6      }
7  };
8  for (i in [1 .. 10]) {
9      for (j in [1 .. 10]) {
10         nPrint(rightAdjust(i * j));
11     }
12     print();
13 }

```

Figure 3.18: A simple program to generate a multiplication table.

1	2	3	4	5	6	7	8	9	10
2	4	6	8	10	12	14	16	18	20
3	6	9	12	15	18	21	24	27	30
4	8	12	16	20	24	28	32	36	40
5	10	15	20	25	30	35	40	45	50
6	12	18	24	30	36	42	48	54	60
7	14	21	28	35	42	49	56	63	70
8	16	24	32	40	48	56	64	72	80
9	18	27	36	45	54	63	72	81	90
10	20	30	40	50	60	70	80	90	100

Figure 3.19: Output of the program in Figure 3.18.

```
for (c in "abc") { print(c); }
```

prints the characters "a", "b", and "c".

The iterator of a **for** loop can also be a *tuple iterator*. The simplest form of a tuple is

$[x_1, \dots, x_n]$ “in” s .

Here, s must be either a set or a list that contains lists of length n as elements. Figure 3.20 on page 36 shows a procedure that computes the relational product of two binary relations r_1 and r_2 . In set theory, the relational product $r_1 \circ r_2$ is defined as

$$r_1 \circ r_2 := \{ \langle x, z \rangle \mid \langle x, y \rangle \in r_1 \wedge \langle y, z \rangle \in r_2 \}.$$

The **for** loop in line 3 iterates over the two relations **r1** and **r2**. The following **if** statement selects those pairs of pairs of numbers such that the second component of the first pair is identical to the first component of the second pair.

Of course, in SETLX the relational product can be computed more easily via set comprehension. Figure 3.21 on page 36 shows an implementation that is based on set comprehension. It is a bit shorter as the test in line 4 of Figure 3.20 is essentially integrated in the definition of the set in line 2 of Figure 3.21. In general, most occurrences of **for** loops can be replaced by equivalent set definitions. In most cases, the resulting code will be both shorter and easier to understand.

Iterators can be even more complex, since the tuples can be nested, so something like

```
for ([x,y],z in s) { ... }
```

```

1  product := procedure(r1, r2) {
2      r := {};
3      for ([x, y1] in r1, [y2, z] in r2) {
4          if (y1 == y2) {
5              r += { [x, z] };
6          }
7      }
8      return r;
9  };

```

Figure 3.20: A program to compute the relational product of two binary relations.

```

1  product := procedure(r1, r2) {
2      return { [x, z] : [x, y] in r1, [y, z] in r2 };
3  };

```

Figure 3.21: Computing the relational product via set comprehension.

is possible. However, as this feature is rarely needed, we won't discuss it further.

A `for` loop creates a local scope for the iteration variable. Therefore, the last line of the program shown in Figure 3.22 prints the message

`x = 1.`

```

1  x := 1;
2  for (x in "abc") {
3      print(x);
4  }
5  print("x = $x$");

```

Figure 3.22: A program illustrating the scope of a `for` loop.

Chapter 4

Exceptions and Backtracking

In the first section of this chapter we will discuss exceptions as a means to deal with error situations. The second section will introduce a mechanism that supports backtracking. This mechanism is quite similar to the exception handling in the first subsection. Indeed, we will see that the backtracking mechanism provided in SETLX is really just a special case of exception handling.

4.1 Exceptions

If we issue the assignment

```
y := x + 1;
```

while the variable `x` is undefined, SETLX reports the following error:

```
Error in "y := x + 1":  
Error in "x + 1":  
'om + 1' is undefined.
```

Here, evaluation of the expression `x + 1` has raised an *exception* as it is not possible to add a number to the undefined value `om`. This exception is then propagated to the enclosing assignment statement. SETLX offers to handle exceptions like the one described above. The mechanism is similar as in *Java* and uses the keywords “`try`” and “`catch`”. If we have a sequence of statements *stmtList* and we fear that something might go wrong with these statements, then we can put the list of statements into a `try/catch`-block as follows:

```
try {  
    stmtList  
} catch (e) { errorCode }
```

If the execution of *stmtList* executes without errors, then the `try/catch`-block does nothing besides the execution of *stmtList*. However, if one of the statements in *stmtList* does raise an exception, then the execution of *stmtList* is aborted and instead the statement *errorCode* is executed.

Typically, exception handling is necessary when processing user input. Consider the program shown in Figure 4.1. The function `findZero` implements the bisection method which can be used to find the zero of a function. The first argument of `findZero` is a function *f*, while the arguments *a* and *b* are the left and right boundary of the interval where the zero of *f* is sought. Therefore, *a* has to be less than *b*. Finally, for the bisection algorithm implemented in the function `findZero` to work, the function *f* needs to have a sign change in the interval $[a, b]$.

The function `askUser` asks the user to input the function *f* together with the left and right boundary of the interval $[a, b]$. The idea is that the user inputs a term describing the function value of *f* for a given value of *x*. For example, in order to compute the zero of the function

$$x \mapsto x * x - 2$$

the user has to provide the string “`x*x-2`” as input to the `read` command in line 3. The string `s` that is input by the user is then converted into the string

`x |-> x*x-2`

in line 4 and, furthermore, the resulting string is parsed and then evaluated. In this way, the variable `f` in line 4 will be assigned the function mapping `x` to `x*x-2` just as if the user had written

`f := x |-> x*x-2;`

in the command line.

```

1  askUser := procedure() {
2      try {
3          s := read("Please enter a function: ");
4          f := evalTerm(parse("x |-> " + s));
5          a := read("Enter left boundary: ");
6          b := read("Enter right boundary: ");
7          z := findZero(f, a, b);
8          print("zero at z = $$");
9      } catch (e) {
10         print(e);
11         print("Please try again.\n");
12         askUser();
13     }
14 };
15 findZero := procedure(f, a, b) {
16     if (a > b) {
17         throw("Left boundary a has to be less than right boundary b!");
18     }
19     [ fa, fb ] := [ f(a), f(b) ];
20     if (fa * fb > 0) {
21         throw("Function f has to have a sign change in [a, b]!");
22     }
23     while (b - a >= 10 ** -12) {
24         c := 1/2 * (a + b);
25         fc := f(c);
26         if ((fa < 0 && fc < 0.0) || (fa >= 0 && fc >= 0)) {
27             a := c; fa := fc;
28         } else {
29             b := c; fb := fc;
30         }
31     }
32     return 1/2 * (a + b);
33 };

```

Figure 4.1: The bisection method for finding a zero of a function.

There are a couple of things that can go wrong with the function `askUser`. First, the string `s` input by the user might not be a proper function and then we would probably get a parse error in line 4. In this case, the function `parse` invoked in line 4 will raise an exception. Next, if the user enters a string of the form

```
x ** y
```

then, since the variable `y` is undefined, the evaluation of the function would raise an exception when SETLX tries to evaluate an expression of the form

```
x ** om.
```

Furthermore, either of the two conditions

$$a < b \quad \text{or} \quad f(a) * f(b) \leq 0$$

might be violated. In this case, we raise an exception in either line 17 or line 21. Since we don't want to abort the program on the occurrence of an exception, the whole block of code in lines 3 up to line 8 is enclosed in a `try/catch`-block. In case there is an exception, the value of this exception, which is a string containing an error message, is caught in the variable `e` in line 9. To continue our program, we print the error message in line 10 and then invoke the function `askUser` recursively so that the user of the program gets another chance.

4.1.1 Different Kinds of Exceptions

SETLX supports two different kinds of exceptions:

1. *User generated* exceptions are generated by the user via a `throw` statement. In general, the statement

```
throw(e)
```

raises a *user generated* exception with the value *e*.

2. *Language generated* exceptions are the result of error conditions arising in the program.

While all kinds of exceptions can be caught with a `catch` clause, most of the time it is useful to distinguish between the different kinds of exceptions. This is supported by offering two variants of `catch`:

1. `catchUsr` only catches user generated exceptions. For example, the statement

```
try { throw(1); } catchUsr(e) { print(e); }
```

prints the number 1, but assuming that the variable `y` is undefined, the statement

```
try { x := y + 1; } catchUsr(e) { print("caught " + e); }
```

will not print anything but instead the command raises an exception.

2. `catchLng` only catches language generated exceptions. Therefore, the statement

```
try { x := y + 1; } catchLng(e) { print("caught " + e); }
```

prints the text

```
caught Error: 'om + 1' is undefined.
```

On the other hand, the exception thrown in the statement

```
try { throw(1); } catchLng(e) { print(e); }
```

is a user generated exception and is therefore not caught.

Being able to distinguish between user generated and language generated exceptions is quite valuable and we strongly advocate that user generated exceptions should only be caught using a `catchUsr` clause. The reason is, that a simple `catch` clause which the user intends to catch a user generated exceptions might, in fact, catch other exceptions and thus mask real errors. In general, the SETLX interpreter implements a *fail fast* strategy: Once an error is discovered, the execution of the program is aborted. The interpreter does a lot of effort to detect errors as early as possible.

However, using unrestricted `catch` clauses counters this strategy and might lead to errors that are very difficult to locate.

4.2 Backtracking

One of the distinguishing features of the programming language *Prolog* is the fact that *Prolog* supports *backtracking*. However, on closer inspection of the *Prolog* programs that are shown in the text books describing *Prolog* [SS86, Bra90] it can be seen that very few programs actually make use of backtracking in its most general form. Also, the personal experience of the first author, who has programmed in *Prolog* for more than 10 years, suggests that *Prolog* programs that use backtracking in an unrestricted fashion tend to be very hard to maintain. In general, it is our believe that the use of backtracking should always follow the paradigm *generate and test*:

1. The set of possible values should be generated by a *generating function*.
2. These values should then be tested. If a test fails, the program backtracks to step 1 where the next value to be tested is generated.

In order to support the generate and test paradigm, SETLX implements backtracking only in a very restricted form. Thus, we avoid the pitfalls that accompany an unrestricted use of backtracking. Backtracking is implemented via the keywords “`check`” and “`backtrack`”. A block of the form

```
check {
    stmtList
}
```

is more or less¹ converted into a block of the form

```
try {
    stmtList
} catch (e) {
    if (e != "fail") {
        throw(e);
    }
}
```

while the keyword “`backtrack`” is translated into “`throw("fail")`”.

The program in Figure 4.2 on page 41 solves the *8 queens puzzle*. This problem asks to position 8 queens on a chessboard such that no queen can attack another queen. In chess, a queen can attack all those positions that are either on the same row, on the same column, or on the same diagonal as the queen. The details of the program in Figure 4.2 are as follows.

1. The procedure `solve` has two parameters.
 - (a) The first parameter `l` is a list of positions of queens that have already been placed on the board. It can be assumed that the queens already positioned in `l` do not attack each other.

Technically, `l` is a list on integers. If `l[i] = k`, then row `i` contains a queen in column `k`. For example,

$$l = [4, 8, 1, 3, 6, 2, 7, 5]$$

is a solution of the 8 queens puzzle. This solution is depicted in Figure 4.3.

- (b) The second parameter `n` is the size of the board.

¹ Technically, instead of the string “`fail`”, SETLX generates a unique exception, which can only be caught using `check`.

```

1  solve := procedure(l, n) {
2      if (#l == n) {
3          return l;
4      }
5      for (x in {1 .. n} - {i : i in l}) {
6          check {
7              testNext(l, x);
8              return solve(l + [x], n);
9          }
10     }
11     backtrack;
12 };
13 testNext := procedure(l, x) {
14     m := #l;
15     if (exists (i in {1 .. m} | i-l[i] == m+1-x || i+l[i] == m+1+x)) {
16         backtrack;
17     }
18 };

```

Figure 4.2: Solving the 8 queens puzzle using `check` and `backtrack`.




Figure 4.3: A solution of the 8 queens puzzle.

In order to solve the 8 queens puzzle, the procedure `solve` is called in line 26 in the form

`solve([], 8)`.

Taking the parameter `l` to be the empty list assumes that initially no queen has been set on the chess board. Of course, then the assumption that the queens already positioned in `l` do not attack each other is trivially satisfied.

2. In line 2 it is checked, whether the list `l` already specifies the positions of `n` queens. If this

is the case, then because of the assumption that the queens specified in `l` do not attack each other, the problem is solved and therefore `l` is the solution and is returned.

3. Otherwise, we find a position `x` for the next queen in line 5. Of course, there is no point in trying to position the next queen into a row that has already been taken by one of the queens in the list `l`. Therefore, the number of positions available for the next queen is given by the set

$$\{1 \dots n\} - \{i : i \text{ in } l\}.$$

Note that we had to convert the list `l` into the set $\{i : i \text{ in } l\}$ in order to be able to subtract the positions specified in `l` from the set of all possible positions.

4. Once we have decided to position the next queen in row `x`, we have to test whether a queen that is put into that position can be attacked by another queen which happens to be on the same diagonal. This test is performed in line 7 with the help of the function `testNext`.
5. If this test succeeds, we add a queen in position `x` to the list `l` and recursively try to solve the resulting instance of the problem.
6. On the other hand, if the call to `testNext` in line 7 fails, we have to try the next value of `x`. Now the function `testNext` does not return a Boolean value to indicate success or failure so at this point you might well ask how we know that the call to check has failed. The answer is that the function `testNext` includes a call to `backtrack` if it is not possible to place a queen in position `x`. Technically, calling `backtrack` raises an exception that is caught by the `check` statement in line 6. After that, the `for` loop in line 5 proceeds and picks the next candidate for `x`.
7. During the recursive invocation of the procedure `solve` in line 8, we might discover that the list `l + [x]` can not be completed into a solution of the 8 queens puzzle. In this case, it is the function `solve` that backtracks in line 11. This happens when the `for` loop in line 5 is exhausted and we have not found a solution. Then control reaches line 11, where the `backtrack` statement signals that the list `l` could not be completed into a solution to the `n` queens puzzle.
8. The function `testNext` in line 13 has two parameters: The first parameter is the list of already positioned queens while the second parameter specifies the column of the next queen. The function checks whether the queen specified by `x` is on the same diagonal as any of the queens in `l`.

In order to understand the calculation in line 15 we have to realize that the cartesian coordinates of the queens in column `x` are

$$\langle \#l + 1, x \rangle.$$

Now a diagonal is specified as the equation of a line with slope either $+1$ or -1 . The i -th queen in `l` has the coordinates

$$\langle i, l[i] \rangle.$$

Therefore, it is on the same ascending diagonal as the queen specified by `x` if we have

$$i - l[i] = \#l + 1 - x,$$

while it is on the same descending diagonal if we have.

$$i + l[i] = \#l + 1 + x.$$

It is easy to change the program in Figure 4.2 such that all solutions are completed. Figure 4.4 on page 43 shows how this is done.

1. We have added a function `allSolutions`. This function gets on parameter `n`, which is the size of the board. The function returns the set of all solutions of the n queens puzzle. To do so, it first initializes the variable `all` to the empty set. The solutions are then collected in this set.
2. The function `solve` gets `all` as an additional parameter. Note that this parameter is specified in line 2 as an `rw` parameter, so the value of `all` is actually changed by the procedure `solve`.
3. The important change in the implementation of `solve` is that instead of returning a solution, a solution that is found is added to the set `all` in line 10. After that, the function `solve` backtracks to look for more solutions.
4. The implementation of the function `testNext` has not changed.

This program finds all 92 solutions to the 8 queens puzzle.

```

1  allSolutions := procedure(n) {
2      all := {};
3      check {
4          solve([], n, all);
5      }
6      return all;
7  };
8  solve := procedure(l, n, rw all) {
9      if (#l == n) {
10         all += { l };
11         backtrack;
12     }
13     for (x in {1 .. n} - {i : i in l}) {
14         check {
15             testNext(l, x);
16             return solve(l + [x], n, all);
17         }
18     }
19     backtrack;
20 };
21 testNext := procedure(l, x) {
22     m := #l;
23     if (exists (i in {1 .. m} | i-l[i] == m+1-x || i+l[i] == m+1+x)) {
24         backtrack;
25     }
26 };

```

Figure 4.4: Computing all solutions of the n queens puzzle.

The keyword `check` can be used with an additional optional branch. In this case the complete `check` block has the form

```

check {
    stmtList
} afterBacktrack { body }

```

Here, *body* is a list of statements that is executed if there is a call to `backtrack` in *stmtList*. For example, the code

```
check {  
    print(1);  
    backtrack;  
    print(2);  
} afterBacktrack {  
    print(3);  
}
```

prints the number 1 and 3.

Chapter 5

Predefined Functions

This chapter lists the predefined functions. The chapter is divided into eight sections.

1. The first section lists all functions that are related to sets and lists.
2. The second section lists all functions that are related to strings.
3. The third sections lists all functions that are used to work with terms.
4. The following sections lists all mathematical functions.
5. The next section list all functions that are used to test whether an object has a given type.
6. Section six lists the functions that support interactive debugging.
7. Section seven discusses the functions related to I/O.
8. The last sections lists all those procedures that did not fit in any of the previous sections.

5.1 Functions and Operators on Sets and Lists

Most of the operators and functions that are supported on sets and lists have already been discussed. However, for the convenience of the reader, this section describes all operators and functions. However, the discussion of those functions that have already been described previously will be quite short.

1. $+$: For sets, the binary operator “ $+$ ” computes the union of its arguments. For lists, this operator appends its arguments.
2. $*$: If both arguments are sets, the operator “ $*$ ” computes the intersection of its arguments. If one argument is a list l and the other argument is a number n , then the list l is appended to itself n times. Therefore, the expression

$[1, 2, 3] * 3$

yields the list

$[1, 2, 3, 1, 2, 3, 1, 2, 3]$

as a result. Instead of a list, the argument l can also be a string. In this case, the string l is replicated n times.

3. $-$: The operator “ $-$ ” computes the set difference of its arguments. This operator is only defined for sets.

4. **%**: The operator “%” computes the symmetric difference of its arguments. This operator is only defined for sets.

Of course, all of the operators discussed so far are also defined on numbers and have the obvious meaning when applied to numbers.

5. **+/**: The operator “+ /” computes the sum of all the elements in its argument. These elements need not be numbers. They can also be sets, lists, or strings. For example, if s is a set of sets, then the expression

$$+ / s$$

computes the union of all sets in s . If s is a list of lists instead, the same expression builds a new list by concatenating all lists in l .

6. ***/**: The operator “*/” computes the product of all the elements in its argument. These elements might be numbers or sets. In the latter case, the operator computes the intersection of all elements.
7. **arb**: The function **arb**(s) picks an arbitrary element from the sequence s . The argument s can either be a set, a list, or a string.
8. **first**: The function **first**(s) picks the first element from the sequence s . The argument s can either be a set, a list, or a string. For a set s , the first element is the element that is smallest with respect to the function **compare** discussed in the last section of this chapter.
9. **last**: The function **last**(s) picks the last element from the sequence s . The argument s can either be a set, a list, or a string. For a set s , the last element is the element that is greatest with respect to the function **compare** discussed in the last section of this chapter.
10. **from**: The function **from**(s) picks an arbitrary element from the sequence s . The argument s can either be a set, a list, or a string. This element is removed from s and returned. This function returns the same element as the function **arb** discussed previously.
11. **fromB**: The function **fromB**(s) picks the first element from the sequence s . The argument s can either be a set, a list, or a string. This element is removed from s and returned. This function returns the same element as the function **first** discussed previously.
12. **fromE**: The function **fromB**(s) picks the last element from the sequence s . The argument s can either be a set, a string, or a list. This element is removed from s and returned. This function returns the same element as the function **last** discussed previously.

13. **domain**: If r is a binary relation, then the equality

$$\text{domain}(r) = \{ x : [x, y] \text{ in } R \}$$

holds. For example, we have

$$\text{domain}(\{[1, 2], [1, 3], [5, 7]\}) = \{1, 5\}.$$

14. **max**: If s is a set or a list, the expression **max**(s) computes the biggest element of s . For example, the expression

$$\text{max}(\{1, 2, 3\})$$

returns the number 3. In general, the elements in s are compared using the function **compare** discussed in the last section of this chapter. For example, strings are compared lexicographically, therefore

$$\text{max}(\{ "abc", "xy", "z" \})$$

yields the result “z”.

15. **min**: If s is a set or a list, the expression `min(s)` computes the smallest element of s . For example, the expression

`min({1,2,3})`

returns the number 1. In general, the elements in s are compared using the function `compare`. For example, strings are compared lexicographically, therefore

`min({"abc", "xy", "z" })`

yields the result `"abc"`.

16. **pow**: If s is a set, the expression `pow(s)` computes the power set of s . The power set of s is defined as the set of all subsets of s . For example, the expression

`pow({1,2,3})`

returns the set

`{{}, {1}, {1, 2}, {1, 2, 3}, {1, 3}, {2}, {2, 3}, {3}}`.

17. **range**: If r is a binary relation, then the equality

`range(r) = { y : [x,y] in R }`

holds. For example, we have

`range({ [1,2], [1,3], [5,7] }) = {2,3,7}`.

18. **reverse**: If l is a list or string, then `reverse(l)` returns a list or string that contains the elements of l in reverse order. For example,

`reverse([1,2,3])`

returns the list

`[3,2,1]`.

19. **sort**: If l is a list or string, then `sort(l)` sorts l into ascending order. For example,

`sort([3,2,1])`

returns the list

`[1,2,3]`.

5.2 Functions for String Manipulation

SETLX provides the following functions that are related to strings.

1. **char**: For a natural number $n \in \{1, \dots, 127\}$, the expression

`char(n)`

computes the character with the ASCII code n . For example, `char(65)` yields the string `"A"`.

2. **endsWith**: The function `endsWith` is called as

`endsWith(s , t)`.

Here, s and t have to be strings. The function succeeds if the string t is a suffix of the string s , i.e. if there is a string r such that the equation

$$s = r + t$$

holds.

3. **eval**: The function **eval** is called as

```
eval(s).
```

Here, s has to be a string that can be parsed as a SETLX expression. This expression is then evaluated in the current variable context and the result of this evaluation is returned. Note that s can describe a SETLX expression of arbitrary complexity. For example, the statement

```
f := eval("procedure(x) { return x * x; }");
```

has the same effect as the following statement:

```
f := procedure(x) { return x * x; };
```

Defining a function f via **eval**(s) is useful because the string s can be the result of an arbitrary computation.

4. **execute**: The function **execute** is called as

```
execute(s).
```

Here, s has to be a string that can be parsed as a SETLX statement. This statement is then executed in the current variable context and the result of this evaluation is returned. Note that s can describe a SETLX expression of arbitrary complexity. For example, the statement

```
execute("f := procedure(x) { return x * x; };");
```

has the same effect as the following statement:

```
f := procedure(x) { return x * x; };
```

5. **join**: The function **join** is called as

```
join(s,t).
```

Here, s is either a set or a list. First, the elements of s are converted into strings. Then these elements are concatenated using the string t as separator. For example, the expression

```
join([1,2,3], "*")
```

yields the string "1*2*3".

The function **join** comes in handy to generate comma separated values.

6. **split**: The function **split** is called as

```
split(s,t).
```

Here, s and t have to be strings. t can either be a single character or a regular expression. The call **split**(s,t) splits the string s at all occurrences of t . The resulting parts of s are collected into a list. If t is the empty string, the string s is split into all of its characters. For example, the expression

```
split("abc", "");
```

returns the list ["a", "b", "c"]. As another example,

```
split("abc xy z", " +");
```

yields the list

```
["abc", "xy", "z"].
```

Note that we have used the regular expression "+" to specify one or more blank characters.

Certain *magic* characters, i.e. all those characters that serve as operator symbols in regular expressions have to be escaped if they are intended as split characters. Escaping is done by prefixing two backslash symbols to the respective character as in the following example:

```
split("abc|xyz", "\\|");
```

The function `split` is very handy when processing comma separated values from CVS files.

7. **str**: The function `str` is called as

```
str(a)
```

where the argument *a* can be anything. This function computes the string representation of *a*. For example, after defining the function `f` as

```
f := procedure(n) { return n * n; };
```

the expression `str(f)` evaluates to the string

```
"procedure(n) { return n * n; }".
```

8. **toLowerCase**: Given a string *s*, the expression `toLowerCase(s)` converts all characters of *s* to lower case.
9. **toUpperCase**: Given a string *s*, the expression `toUpperCase(s)` converts all characters of *s* to upper case.
10. **trim**: Given a string *s*, the expression `trim(s)` returns a string that is *s* without all leading or trailing white space characters. For example, the expression

```
trim(" abc xyz\n")
```

returns the string "abc xyz".

5.3 Functions for Term Manipulation

The following functions support terms.

1. **args**: Given a term *t* that has the form

$$F(s_1, \dots, s_n),$$

the expression `args(t)` returns the list

$$[s_1, \dots, s_n].$$

2. **evalTerm**: This function is called as

```
evalTerm(t).
```

Here, *t* has to be a term that represents a SETLX expression. This expression is then evaluated in the current variable context and the result of this evaluation is returned. For convenience, the term *t* can be produced by the function `parse`. For example, the statement

```
f := evalTerm(parse("procedure(x) { return x * x; }"));
```

has the same effect as the following statement:

```
f := procedure(x) { return x * x; };
```

The function `evalTerm` is an advanced feature of SETLX that allows for self modifying programs. This idea is that a function definition given as a string can be transformed into a term. This term can then be manipulated using the facilities provided by the `match` statement and the modified term can finally be evaluated using `evalTerm`.

3. **fct**: Given a term *t* that has the form

$$F(s_1, \dots, s_n),$$

the expression `fct(t)` returns the functor *F*.

4. **getTerm**: The function **getTerm** is called as

```
getTerm(v).
```

It returns a term representing the value *v*. For example, the expression

```
canonical(getTerm(procedure(n) { return n * n; })))
```

produces the following output:

```
^procedure([ ^parameter(^variable("n"))],
            ^block([ ^return(^product(^variable("n"), ^variable("n")))]))
)
```

However, there is one twist: If *v* is either a set or a list, then **getTerm**(*v*) transforms the elements of *v* into terms and returns the resulting set or list.

5. **makeTerm**: Given a functor *F* and a list *l* = [*s*₁, ..., *s*_{*n*}], the expression

```
makeTerm(F, [s1, ..., sn])
```

returns the term

```
F(s1, ..., sn).
```

6. **canonical**: Given a term *t*, the expression **canonical**(*t*) returns a string that is the canonical representation of the term *t*. The point is, that all operators in *t* are replaced by functors that denote these operators internally. For example, the expression

```
canonical(parse("x+2*y"));
```

yields the string

```
^sum(^variable("x"), ^product(2, ^variable("y"))).
```

This shows that, internally, variables are represented using the functor **^variable** and that the operator “+” is represented by the functor **^sum**.

7. **parse**: Given a string *s*, the expression

```
parse(s)
```

tries to parse the string *s* into a term. In order to visualize the structure of this term, the function **canonical** discussed above can be used.

8. **parseStatements**: Given a string *s*, the expression

```
parseStatements(s)
```

tries to parse the string *s* as a sequence of SETLX statements. In order to visualize the structure of this term, the function **canonical** discussed above can be used. For example, the expression

```
canonical(parseStatements("x := 1; y := 2; z := x + y;"));
```

yields the following term (which has been formatted for easier readability):

```
^block([ ^assignment(^variable("x"), 1),
        ^assignment(^variable("y"), 2),
        ^assignment(^variable("z"), ^sum(^variable("x"), ^variable("y")))
      ])
```

5.4 Mathematical Functions

The function SETLX provides the following mathematical functions.

1. The operators “+”, “-”, “*”, “/”, and “%” compute the sum, the difference, the product, the quotient, and the remainder of its operands. The remainder $a\%b$ is defined so that it satisfies

$$0 \leq a\%b \quad \text{and} \quad a\%b < b.$$

2. The operator “\” computes integer division of its arguments. For two integers a and b , this is defined such that

$$a = (a \setminus b) * b + a \% b$$

holds.

3. `abs(x)` calculates the absolute value of the number x .
4. `ceil(x)` calculates the *ceiling* function of x . Mathematically, `ceil(x)` is defined as the smallest integer that is bigger than or equal to x :

$$\text{ceil}(x) := \min\{n \in \mathbb{Z} \mid x \leq n\}.$$

For example, we have

$$\text{ceil}(2.1) = 3.$$

5. `floor(x)` calculates the *flooring* function of x . Mathematically, `floor(x)` is defined as the largest integer that is less than or equal to x :

$$\text{floor}(x) := \max\{n \in \mathbb{Z} \mid n \leq x\}.$$

For example, we have

$$\text{floor}(2.9) = 2.$$

6. `mathConst(name)` can be used to compute mathematical constants. At the moment, π and Euler’s number e are supported. Therefore, the expression

`mathConst("pi")`

yields the result 3.141592653589793, while

`mathConst("e")`

yields 2.718281828459045.

7. `nextProbablePrime(n)` returns the next probable prime number that is greater than n . Here, n needs to be a natural number. The probability that the result of `nextProbablePrime(n)` is not a prime number is less than 2^{-100} . For example, to find the smallest prime number greater than 1000 we can use the expression

`nextProbablePrime(1000).`

8. `int(s)` converts the string s into a number. The function `int` can also be called if s is already a number. In this case, if s is an integer number, it is returned unchanged. Otherwise, the floating point part is truncated. Therefore, the expression

`int(2.9)`

returns the integer 2.

9. `rational(s)` converts the string s into a rational number. For example the expression

`rational("2/7")`

will return the rational number $2/7$. The function `rational` can also be invoked on real

numbers. For example, the expression

```
rational(mathConst("e"))
```

yields the result 543656365691809/200000000000000. The expression `rational(q)` returns q if q is already a rational number.

10. `real(s)` converts the string s into a floating point number. For example the expression

```
rational("1.2")
```

will return the floating point number 1.2. The function `real` can also be invoked on rational numbers. For example, the expression

```
real(2/7)
```

yields 0.2857142857142857. The expression `real(r)` returns r if r is already a floating point number.

Warning: SETLX implements floating point numbers via the Java class `BigDecimal` instead of using the primitive data type `double`. Therefore, floating point operations are noticeably slower than in other programming languages.

11. Furthermore, the trigonometrical functions `sin`, `cos`, `tan` and the associated inverse trigonometrical functions `asin`, `acos`, and `atan` are all supported. For example, the expression

```
sin(mathConst("pi")/2);
```

yields 1.0.

12. `exp(x)` computes e^x where e is Euler's number. Therefore, the expression

```
exp(1)
```

yields 2.718281828459046.

13. `log(x)` computes the natural logarithm of x . This function is the inverse function of the exponential function `exp`. Therefore, we have the equation

$$\log(\exp(x)) = x.$$

This equation is valid as long as there is no overflow in the computation of `exp(x)`.

14. `log10(x)` computes the base 10 logarithm of x .

15. `sqrt(x)` computes the square root \sqrt{x} . Therefore, as long as x is not too large, the equation

$$\text{sqrt}(x*x) = x$$

is valid.

16. `cbrt(x)` computes the cubic root $\sqrt[3]{x}$. Therefore, as long as x is not too large, the equation

$$\text{cbrt}(x*x*x) = x$$

is valid.

17. `round(x)` rounds the number x to the nearest integer.

18. `nDecimalPlaces(q , n)` takes a rational number q and a positive natural number n as arguments. It converts the rational number q into a string that denotes the value of q in decimal floating point notation. This string contains n digits after the decimal point. Note that these digits are truncated, there is no rounding involved. For example, the expression

```
nDecimalPlaces(2/3,5)
```

yields the string "0.66666", while

`nDecimalPlaces(1234567/3,5)` returns “411522.33333”.

19. `ulp(x)` returns the difference between the floating point number x and the smallest floating point number bigger than x . For example, when working with 64 bits floating point numbers (which is the default), we have

$$\text{ulp}(1.0) = 2.220446049250313\text{E-}16.$$

`ulp` is the abbreviation for *unit in the last place*.

20. `signum(x)` computes the sign of x . If x is positive, the result is 1.0, if x is negative, the result is -1.0 . If x is zero, `signum(x)` is also zero.
21. `sinh(x)` computes the hyperbolic sine of x . Mathematically, the hyperbolic sine is defined as

$$\sinh(x) := \frac{1}{2} \cdot (e^x - e^{-x}).$$

22. `cosh(x)` computes the hyperbolic cosine of x . Mathematically, the hyperbolic cosine is defined as

$$\cosh(x) := \frac{1}{2} \cdot (e^x + e^{-x}).$$

23. `tanh(x)` computes the hyperbolic tangent of x . Mathematically, the hyperbolic tangent is defined as

$$\tanh(x) := \frac{\sinh(x)}{\cosh(x)}.$$

24. `isPrime` tests whether its argument is a prime number. Currently, this function has a naive implementation and is therefore not efficient.
25. `isProbablePrime` tests whether its argument is a prime number. The test used is probabilistic. If the function tested is indeed prime, the test will succeed. If the function is not prime, the predicate `isProbablePrime` nevertheless might return `true`. However, the probability that this happens is less than 2^{-30} .

It should be noticed that the implementation of `isProbablePrime` is based on random numbers. Therefore, in rare cases different invocations of `isProbablePrime` might return different results.

5.5 Generating Random Numbers and Permutations

In order to generate random numbers, SETLX provides the functions `rnd` and `random`. The easiest to use of these function is `random`. The expression

`random()`

generates a random number x such that

$$0 \leq x \quad \text{and} \quad x \leq 1$$

holds. Normally, this will be a floating point number with 64 bits. If SETLX is instead invoked with the either of the parameters “`--real32`”, “`--real128`”, or “`--real256`” then x will have 32 bits, 128 bits, or 256 bits respectively. For debugging purposes it might be necessary to start SETLX with the parameter

`--predictableRandom`.

In this case, the sequence of random numbers that is generated by the function `random` and `rnd` is always the same. This is useful when debugging a program working with random numbers. The function `random` can also be called with an argument. The expression

`random(n)`

is equivalent to the expression

`n * random()`.

The function `rnd` has a number of different uses.

1. If *l* is either a list or a set, then the expression

`rnd(l)`

returns a random element of *l*.

2. If *n* is an natural number, then

`rnd(n)`

returns a natural number *k* such that we have

$$0 \leq k \text{ and } k \leq n.$$

3. If *n* is a negative number, then

`rnd(n)`

returns a negative number *k* such that we have

$$n \leq k \text{ and } k \leq 0.$$

4. In order to generate random rational numbers, the function `rnd` has to be called with two arguments. The expression

`rnd(a/b, n)`

is internally translated into the expression

`rnd(n - 1) * a / (b * (n - 1)).`

Therefore, if *a* is positive, the number returned is a random number between 0 and the fraction *a/b* and there will be *n* different possibilities, so the parameter *n* is used to control the granularity of the results. For example, the expression

`rnd(1/2,6)`

can return any of the following six fractions:

0, 1/10, 2/10, 3/10, 4/10, 5/10,

while the expression `rnd(1/2,101)` could return 101 different results, namely any fraction of the form

a/100 such that *a* ∈ {0, 1, ..., 100}.

Of course, the user will never see a result of the form 50/100 as this is immediately reduced to 1/2.

The second parameter also works if the first argument is a natural number. For example, the expression `random(1,11)` returns one of the following eleven fractions:

0, 1/10, 2/10, 3/10, 4/10, 5/10, 6/10, 7/10, 8/10, 9/10, 10/10.

5.5.1 shuffle

The function `shuffle` shuffles a list or string randomly. For example, the expression

`shuffle("abcdef")`

might yield the result "dfbaec", while

```
shuffle([1,2,3,4,5,6,7])
```

might yield the list [1, 7, 4, 6, 3, 5, 2].

5.5.2 nextPermutation

The function `nextPermutation` can be used to enumerate all possible permutations of a given list or string. For example, the function `printAllPermutations` shown in Figure 5.1 prints all permutations of a given list or string. Evaluation of the expression

```
printAllPermutations([1,2,3])
```

yields the following output:

```
[1, 2, 3]
[1, 3, 2]
[2, 1, 3]
[2, 3, 1]
[3, 1, 2]
[3, 2, 1]
```

```
1  printAllPermutations := procedure(l) {
2      while (l != om) {
3          print(l);
4          l := nextPermutation(l);
5      }
6  };
```

Figure 5.1: Printing all permutations of a given list or string.

5.5.3 permutations

Given a set, list or string l , the expression

```
permutations(l)
```

returns a set of all permutations of l . For example, the expression

```
permutations("abc")
```

returns the set

```
{"abc", "acb", "bac", "bca", "cab", "cba"}.
```

Likewise, the expression

```
permutations({1, 2, 3})
```

yields the result

```
{[1, 2, 3], [1, 3, 2], [2, 1, 3], [2, 3, 1], [3, 1, 2], [3, 2, 1]}.
```

5.6 Type Checking Functions

SETLX provides the following functions to check the type of a given function.

1. `isBoolean` tests whether its argument is a Boolean value.

2. `isError` tests whether its argument is an error object.

For example, consider the following code:

```
try { throw("foo"); } catch (e) { print(isError(e)); }
```

Here, the exception e that gets caught has been thrown by the user and therefore is not an error. Hence, this program fragment prints `false`. However, the program fragment

```
try { om+1; } catch (e) { print(isError(e)); }
```

will indeed print `true`, as the evaluation of `om+1` raises an exception that signifies an error.

3. `isInfinite` tests whether its argument is a real number that represents either positive or negative infinity.
4. `isInteger` tests whether its argument is an integer number.
5. `isList` tests whether its argument is a list.
6. `isMap` tests whether its argument is a binary relation that maps every element in its domain *uniquely* into an element of its range. Therefore, a binary relation r is not a map if it contains two pairs

$$[x, y_1] \quad \text{and} \quad [x, y_2]$$

such that $y_1 \neq y_2$.

7. `isNumber` tests whether its argument is a number.
8. `isProcedure` tests whether its argument is a procedure.
9. `isRational` tests whether its argument is a rational number.
10. `isReal` tests whether its argument is a floating point number.
11. `isSet` tests whether its argument is a set.
12. `isString` tests whether its argument is a string.
13. `isTerm` tests whether its argument is a term.

5.7 Interactive Debugging

SETLX supports a command line interface for interactive debugging. The interface to the debugger consists of the following functions:

1. `trace`: This function is called as either

```
trace(true)   or   trace(false).
```

The expression `trace(true)` switches tracing on: Afterwards, all assignments are written to the terminal window running SETLX. To switch tracing off, call `trace(false)`.

2. `setBreak`: The expression

```
setBreak(f)
```

sets a breakpoint for the function specified by the argument f . In the context of `setBreak`, the argument f is a string that is the name of a function.

3. **rmBreak**: The expression

`rmBreak(f)`

removes the breakpoint for the function *f*, where *f* is the name of a function.

4. **lsBreak**: The expression

`lsBreak()`

lists all breakpoints.

5. **step**: In general, the expression

`step()`

executes the next statement. However, if the next statement evaluates a complex function, the `step()` only enters this function so that the programmer can then step through the statements of the called function.

6. **uStep**: The expression

`uStep()`

evaluates the expression that is needed to evaluate the next statement. This is called a *micro step*.

7. **fStep**: The expression

`fStep()`

executes a *macro step*: this executes the next instruction. If the next instruction is a function call, then this function call is evaluated in one step, so this function steps over function calls.

8. **finish**: The expression

`finish()`

finishes execution of the current function. This function is handy if you accidentally step into a function.

9. **finishLoop**: The expression

`finishLoop()`

finishes the execution of the current loop without halting.

10. **resume**: The expression

`resume()`

resumes normal execution of the program.

11. **reset**: The expression

`reset()`

stops the execution of the current program and returns to the interactive prompt of the interpreter.

Finally the function `dbgHelp()` lists all commands that are available to support interactive debugging.

5.8 I/O Functions

This section lists all functions related to input and output.

5.8.1 appendFile

The function `appendFile` is called as

```
appendFile(fileName, l).
```

Here, *fileName* is a string denoting the name of a file, while *l* is a list of strings. If file *a* with the specified name does not exist, it is created. Then the strings given in *l* are appended to the file. Each string written to the specified file is automatically terminated by a newline character.

5.8.2 deleteFile

The function `deleteFile` is called as

```
deleteFile(fileName).
```

Here, *fileName* is a string denoting the name of a file. If a file with the specified name does exist, it is deleted. In this case the function returns `true`. If a file with the specified name does not exist, the function returns `false` instead.

5.8.3 get

The function `get` is called as

```
get(s).
```

Here, *s* is a string that is used for a prompt. This argument is optional. The function prints *s* and then returns the string that the user has supplied. If no string is supplied, `get` uses the prompt `": "`.

5.8.4 load

The function `load` is called as

```
load(file).
```

Here, *file* has to be a string that denotes a file name, including the extension of the file. Furthermore, *file* is expected to contain valid SETLX commands. These commands are then executed. In general, most of the commands will be definitions of functions. These function can then be used interactively.

5.8.5 loadLibrary

The function `loadLibrary` is called as

```
loadLibrary(file).
```

Here, *file* has to be a string that denotes a file name, excluding the extension of the file. This file is assumed to be located in the directory that is specified by the environment variable

```
SETLX_LIBRARY_PATH.
```

There are three different ways to set this variable.

1. The variable can be set a in file like `“.profile”` or `“.bashrc”`, or something similar. Depending on the type of shell you are using, these files are automatically executed when a new shell is started.
2. The variable can be set manually in the shell.
3. The variable can be set using the option `“--libraryPath”`.

The specified *file* is expected to contain valid SETLX commands.

5.8.6 multiLineMode

The function `multiLineMode` is called as

```
multiLineMode(flag).
```

Here, `flag` should be an expression that evaluates to a Boolean value. The function either activates or deactivates *multi line mode*. If multi line mode is activated, then in a shell the next input expression is only evaluated after the user hits the return key twice. Multi line mode makes it possible to enter statements spanning several lines interactively. However, normally this mode is inconvenient, as it requires the user to press the return key twice in order to evaluate an expression. Therefore, by default multi line mode is not active. Multi line mode can also be activated using the option “`--multiLineMode`” when starting SETLX.

5.8.7 nPrint

The function `nPrint` is called as

```
nPrint(a1, ..., an).
```

It takes any number of arguments and prints these arguments onto the standard output stream. In contrast to the function `print`, this function does not append a newline to the printed output.

5.8.8 nPrintErr

The function `nPrintErr` is called as

```
nPrintErr(a1, ..., an).
```

It takes any number of arguments and prints these arguments onto the standard error stream. In contrast to the function `printErr`, this function does not append a newline to the printed output.

5.8.9 print

The function `print` is called as

```
print(a1, ..., an).
```

It takes any number of arguments and prints these arguments onto the standard output stream. After all arguments are printed, this function appends a newline to the output.

5.8.10 printErr

The function `printErr` is called as

```
printErr(a1, ..., an).
```

It takes any number of arguments and prints these arguments onto the standard error stream. After all arguments are printed, this function appends a newline to the output.

5.8.11 read

The function `read` is called as

```
read(s).
```

Here, `s` is a string that is used for a prompt. This argument is optional. The function prints `s` and then returns the string that the user has supplied. However, leading and trailing white space is removed from the string that has been read. If the string can be interpreted as a number, this number is returned instead. Furthermore, this function keeps prompting the user for input until the user enters a non-empty string.

5.8.12 readFile

The function `readFile` is called as

```
readFile(file, lines).
```

The second parameter *lines* is optional. It reads the specified file and returns a list of strings. Each string corresponds to one line of the file. This second parameter *lines* specifies the list of line numbers to read. For example, the statement

```
read("file.txt", [42] + [78..113])
```

will read line number 42 and the lines 78 up to and including line 113 of the given file. This feature can be used to read a file in chunks of 1000 lines as in the following example:

```
n := 1;
while (true) {
  content := readFile("file", [n .. n + 999]);
  if (content != []) {
    // process 1k lines
    ....
    n += 1000;
  } else {
    break;
  }
}
```

5.8.13 writeFile

The function `writeFile` is called as

```
writeFile(f, l).
```

Here, *f* is the name of the file and *l* is a list. The file *f* is created and the list *l* is written into the file *f*. The different elements of *l* are separated by newlines.

5.9 Miscellaneous Functions

This final section lists some functions that did not fit into any of the other sections.

5.9.1 abort

This function aborts the execution of the current function. This is done by raising an exception. Usually, the function `abort` is called with one argument. This argument is then thrown as an exception. For example, the statement

```
try { abort(1); } catch (e) { print("e = e"); }
```

will print the result

```
e = Error: abort: 1.
```

Note that an exception raised via `abort` can not be caught with the keyword `catchUsr`. The keyword `catchUsr` will only catch exceptions that are explicitly thrown by the user via invocation of `throw`.

5.9.2 cacheStats

The function `cacheStats` is called as

`cacheStats(f).`

Here, f is a cached procedure, i.e. a procedure that is declared using the keyword “`cachedProcedure`”. Note that f has to be the procedure itself, not the name of the procedure! The returned result is a set of the form

`{["cache hits", 996], ["cached items", 1281]}.`

This set can be interpreted as a map.

5.9.3 clearCache

The function `clearCache` is called as

`clearCache(f).`

Here, f is a cached procedure, i.e. a procedure that is declared using the keyword “`cachedProcedure`”. Note that f has to be the procedure itself, not the name of the procedure! The invocation of `clearCache(f)` frees the memory associated with the cache for the function f . It should be used if the previously computed values of f are not likely to be needed for the next computation.

5.9.4 compare

The function `compare` is called as

`compare(x, y).`

Here, x and y are two arbitrary values. This function returns -1 if x is less than y , $+1$ if x is bigger than y and 0 if x and y are equal. If x and y have a numerical type, then the result of `compare(x, y)` coincides with the result produced by the operator “ $<$ ”. If x and y are both lists, then the lists are compared lexicographically. The same remark holds if x and y are both sets. If x and y have different types, then the result of `compare(x, y)` is implementation defined. Therefore, the user should not rely on the results returned in these cases.

The function `compare` is needed internally in order to compare the elements of a set. In SETLX all sets are represented as ordered binary trees.

5.9.5 getScope

The function `getScope` is called as

`getScope().`

It returns a term representing the *current scope*. Here, the current scope captures the binding of all variables. For example, suppose the user issues the following commands:

```
x := 1;
y := 2;
f := procedure(n) { return n * n; };
```

Then the current scope consists of the variables x , y , and f . Therefore, in this case the expression `getScope()` returns the following term:

```
^scope({["f", procedure(n) { return n * n; }],
        ["getScope", ^preDefinedProcedure("getScope")], ["params", []],
        ["x", 1],
        ["y", 2]
      })
```

5.9.6 `logo`

The function `logo` is called as

`logo()`.

In order to find how this function works, try it yourself.

5.9.7 `now`

The function `now` returns the number of milliseconds that have elapsed since the beginning of the Unix epoch.

5.9.8 `sleep`

The function `sleep` takes one argument that has to be a positive natural number. The expression

`sleep(t)`

pauses the execution of the process running SETLX for t milliseconds. When printing output, this can be used for visual effects.

Bibliography

- [Ada80] Douglas Adams. *The Hitchhiker's Guide to the Galaxy*. Harmony Books, New York, 1980.
- [Bra90] Ivan Bratko. *PROLOG Programming for Artificial Intelligence*. Addison-Wesley, 2nd edition, 1990.
- [FM08] David Flanagan and Yukihiro Matsumoto. *The Ruby Programming Language*. O'Reilly, 2008.
- [Sch70] Jacob T. Schwartz. Set theory as a language for program specification and programming. Technical report, Courant Institute of Mathematical Sciences, New York University, 1970.
- [SDSD86] J. T. Schwartz, R. B. Dewar, E. Schonberg, and E. Dubinsky. *Programming With Sets: An Introduction to SETL*. Springer-Verlag, 1986.
- [Sny90] W. Kirk Snyder. The Setl2 programming language: Update on current developments. Technical report, Courant Institute of Mathematical Sciences, New York University, 1990. Available at <ftp://cs.nyu.edu/pub/languages/setl2/doc/2.2/update.ps>.
- [SS86] Leon Sterling and Ehud Shapiro. *The Art of Prolog*. MIT Press, 1986.
- [vR95] Guido van Rossum. Python tutorial. Technical report, Centrum Wiskunde & Informatica, Amsterdam, 1995.